
6 Counting

Introduction
This chapter presents a variety of techniques that are available in Maple for counting a diverse 
collection of discrete objects, including combinations and permutations of finite sets.  Objects can be
counted using formulae or by using algorithms to list the objects and then directly count the size of 
the list. 

Most of the Maple commands relevant to this chapter dwell in the combinat package.  Since this 
package will be used extensively in this chapter, we load it now.

with(combinat):

Advanced readers may wish to explore the combstruct package on their own.  This package will
not be discussed in this manual as the combinat package is sufficient for the material covered 
here.  The combstruct package contains commands related to combinatorial structures, and can 
be thought of as generalizing some of the commands provided by combinat.

6.1 The Basics of Counting
In this section we will see how Maple can be used to perform the computations needed to solve 
basic counting problems.  We will begin by looking at some examples.  We will discuss 
computations involving large integers.  Then we'll see how the principles of counting can be used to
count the number of operations used by a Maple procedure.  This section concludes by using Maple
procedures to solve counting problems by enumerating all the possibilities.

Basic Examples
We begin with two basic examples to demonstrate the use of some useful Maple commands.  

Counting one-to-one functions
Recall Example 7 from Section 6.1 of the text.  This example calculated that the number of one-to-
one functions from a set with m elements to a set with n elements is 

n nK 1 nK 2 / nKm C 1 .
Note that we can rewrite this using product notation as 

?
i = 0

mK 1

nK i.

For small values of m, it is easy to enter this product in Maple.  For instance, the statement below 
computes the number of one-to-one functions from a set of 4 elements to a set of 20 elements.

20*19*18*17;
116280

For larger values of m, it is more convenient to use the mul command.  The mul command is used 
to multiply a sequence of values.  Its syntax is similar to the syntax for seq.  The first argument is 
an expression in terms of a variable (e.g., i) that evaluates to the values that are to be multiplied 
together.  The second argument, in the most common usage, has the form i=a..b with the range
a..b representing the values the variable is to take.

For example, we can recompute the number of one-to-one functions from a set of 20 elements to a 

set of 4 elements as follows.
mul(20-i,i=0..3);

116280
The second argument indicates that the index variable i will be assigned the integers 0, 1, 2 and 3.  
These are then substituted into the first argument, 20-i, producing the values 20, 19, 18, and 17, 
which are multiplied together.

We can easily compute the number of one-to-one functions from a set of 12 elements to a set of 300
elements.

mul(300-i,i=0..11);
425270752192695317567218560000

Computer passwords
Example 16 from Section 6.1 describes a computer system in which each user has a password that 
must be between six and eight characters long consisting of uppercase letters or digits.  Also, each 
password must contain at least one digit.

The solution to the example describes how to calculate this.  For each password length, 6, 7, or 8, 
the number of passwords are

P[6] := 36^6 - 26^6;
P6 := 1867866560

P[7] := 36^7 - 26^7;
P7 := 70332353920

P[8] := 36^8 - 26^8;
P8 := 2612282842880

Thus the total number of possible passwords is
P := P[6] + P[7] + P[8];

P := 2684483063360

We can use the add command to make this calculation a bit easier.  The add command has the 
same syntax as mul, but is used to add its arguments rather than multiply them.

add(36^i - 26^i,i=6..8);
2684483063360

This makes it easier to compute the number of valid passwords for larger ranges.  For instance, it is 
common to require passwords to be between 8 and 16 characters.  If we retain the rules that the 
password be uppercase letters or numbers and include at least one number, then the total number of 
possible passwords is calculated with the statement below.

add(36^i - 26^i,i=8..16);
8140698334757962630580480

Working with Large Integers
Maple's computational engine is able to work with arbitrarily large integers, subject only to the 
limitations imposed by the computer's memory and speed.  

DNA
In Example 11, the text provides a brief description of DNA and concludes that there are at least 

4108
 different possible sequences of bases in the DNA for complex organisms.  

To have Maple compute this value, we just enter the statement.  (Note that since exponentiation is 
not associative, parentheses are required in this calculation.)

DNAsequences := 4^(10^8);
DNAsequences := ...Integer too large for display...

Maple reports that the result is "too large for display."  This does not imply that Maple has not 
computed it, it only means that displaying the integer would require excessive space.  

Maple has computed and stored the exact value and we can use this value in further computations.  
For example, we can find the last three digits of the number by computing the result modulo 1000.

DNAsequences mod 1000;
376

We can calculate the number of digits in the result by applying the ilog10 command.  This 
command computes an integer approximation of the base 10 logarithm of its argument.  Specifically,
as long as the argument x is a real number, the result of ilog10(x) is an integer r such that 
10r ! x ! 10r C 1.  In particular, the result is one less than the number of digits of x.

Let's apply ilog10 to the result for the number of possible sequences of bases in DNA.
ilog10(DNAsequences);

60205999

Remember that the approximation 4108
 was a lower bound.  In other words, the minimum number 

of possible sequences of bases in the DNA of a complex organism has over 60 million digits.

Suppose you wanted to print this number.  Using a typical fixed-width 12-point font and 1-inch 
margins, you can fit about 64 digits in each line and 45 lines on a page.  With these parameters, it 
would require

evalf((6.12)/(64*45));
20904.86076

pages to print the entire number.

Variable names in Maple
Example 15 in Section 6.1 of the text calculated that in one version of the programming language 
BASIC, there were 957 different names for variables.  We will calculate the number of possible
names in Maple.  

In Maple, the basic form of a name is a letter possibly followed by additional letters, digits or 
underscores.  Maple considers uppercase distinct from lowercase.   There are 52 uppercase or 
lowercase letters that can be used as the first character.  And, with the underscore and the 10 digits, 
there are 63 possibilities for each character following the first.

The maximum length of a name depends on the computer.  On a 32-bit platform, the maximum 
length is 524,271 characters.  On a 64-bit machine, the maximum is over 34 billion characters long. 
It is unlikely that you would ever use such a name, but you could if you wanted to.

How many possible names are there?  Considering only the names of maximum length on a 32-bit 
machine, the number of possibilities is computed below.  (Note that the total number of possible 

names is >
i = 0

524270

52$63i.)

52*63^524270;
967977746018687422271630340478755603105726912227078884029485090720903770\

5300451030827278793691911420[...943142 digits...]
781029241857201161514662124392550878960153533571843278323927226147758
8672669963929287534682879863348

This time Maple displayed part of the result, specifically the first 100 and last 100 digits.  In 
between, it told us that it omitted 943,142 digits in displaying the result.  

We will ask a more reasonable question.  How many names have at most 15 characters?  To answer
this question, we'll use the add command, as we did above.  The formula will be 52$63i where the 
index variable i represents the number of characters in the name beyond the first and ranges from 0 
to 14.

add(52*63^i,i=0..14);
819822618169347817599853876

We see that even limiting ourselves to a maximum of 15 characters, there are over 800 septillion 
distinct Maple names.  (The number above is slightly inaccurate because it does not exclude Maple 
keywords and other protected names that you are not allowed to assign values to.  Of course, those 
are relatively insignificant.)

Counting Operations in a Procedure
Next we'll consider an example of counting the number of operations performed by a procedure.  In 
Example 9 in Section 6.1 of the textbook, it is shown that the number of times that the innermost 
statement in a nested for loop is executed is the product of the number of iterations of each loop.

As an example of this, we'll consider the MakePostage procedure from Section 5.1 of this 
manual.  Recall that the purpose of this procedure was to determine the numbers of stamps of two 
given denominations that are required to make a given amount of postage.  Here is the procedure 
definition again.

MakePostage := proc(A::posint,B::posint,postage::posint)
  local a, b;
  for a from 0 to floor(postage/A) do
    for b from 0 to floor(postage/B) do
      if A*a + B*b = postage then
        return [a,b];
      end if;
    end do;
  end do;
  return FAIL;
end proc:

We will count the number of multiplications and additions that this procedure requires in the worst 
case.  The return command means that once the procedure has found a way to make the desired 
amount of postage, execution is immediately terminated.  This means that knowing the number of 
iterations used for a particular input value is equivalent to knowing the output of the procedure.  If 
there was a formula, we wouldn't need the procedure.  By considering the worst-case scenario, we 
can get an idea of the complexity of the algorithm without having to execute the procedure.



O O 
O O 

(6.16)(6.16)

The worst-case scenario, that is, the situation that requires the most number of iterations of the loop,
occurs when the desired postage cannot be made.  In this case, the outer loop variable will range 

from 0 to 
postage

A
 and the inner loop will range from 0 to 

postage
B

.  Thus the number of times 

the if statement is executed is 
postage

A
C 1

postage
B

C 1 .  Therefore, in the worst case, 

the MakePostage procedure requires that number of additions and twice as many multiplications.

Counting by Listing All Possibilities
At the end of Section 6.1, the text discusses using tree diagrams to solve counting problems.  Tree 
diagrams provide a visual way to organize information so you can be sure that you arrive at all 
possible results.  We will not, in this section, implement trees, as that is the focus of Chapter 11.  
The goal of a tree diagram is to list all of the possibilities.  In this subsection we will consider two 
problems that can be solved by using Maple procedures to list all possibilities.  

Subsets
For the first example, we consider the following question: how many subsets of the set of the 
integers 1 through 10 have sums less than 15?  (This is similar to Exercise 67 in Section 6.1.)

To solve this problem, we'll consider all of the possible subsets and count those that satisfy the 
condition.  

In order to generate all of the possible subsets of 1, 2,…, 10 , we'll use the subsets command, 
first introduced in Section 2.1 of this manual.  The subsets command is part of the combinat 
package.  

The subsets command accepts one argument: the set (or list) whose subsets are to be generated.  
It returns a table with two entries.  The nextvalue entry is a procedure that takes no arguments 
and that, when executed, returns a subset.  The finished entry is a boolean value that is initially 
false but is set to true once the nextvalue procedure has returned the final subset.

Here is an example of using the subsets command to print all of the subsets of the set 1, 2 .  
subs12 := subsets({1,2}):
while not subs12[finished] do
  print(subs12[nextvalue]());
end do;

1
2

1, 2
The assignment to the result of subsets establishes subs12 as the name storing the table.  Then
subs12[finished] accesses the boolean value of the finished entry from the table.  As this 
is false until all subsets have been produced, its logical negation is a useful control for a while loop. 
To produce the subsets, subs12[nextvalue]() calls the nextvalue procedure, which 
returns the "next" subset.  

We will use subsets to solve the problem of counting the number of subsets of 1, 2, 3,…, 10  
whose sum is less than 15.  Instead of printing all subsets as the simple loop above does, we'll 
instead test the subset produced by nextvalue to see if its sum is less than 15.

To calculate the sum of the elements of a set, we'll use the add command.  Recall that the add 
command allows the second argument to include a list or set of values instead of a range.  For 
example, to add the elements of the set 1, 3, 7, 8 , we can execute the following expression. 

add(i,i={1,3,7,8});
19

Since the problem, as stated, was to count the number of subsets with sum less than 15, we'll use a 
variable that is initialized to 0 and incremented each time a subset has the desired property.

Also, we'll generalize a bit and write a procedure that accepts a set of positive integers and a target 
value and counts all the subsets of the given set whose elements sum to a value less than the target. 

SubsetSumCount := proc(S::set(posint),target::posint)
  uses combinat;
  local count, P, s, x;
  count := 0;
  P := subsets(S);
  while not P[finished] do
    s := P[nextvalue]();
    x := add(i,i=s);
    if x < target then
      count := count + 1;
    end if;
  end do;
  return count;
end proc:

Applying the procedure to 1, 2, 3,…, 10  and 15 will answer the original question.
SubsetSumCount({1,2,3,4,5,6,7,8,9,10},15);

99

Bit strings
For the second example, we'll consider a problem similar to Example 21.  How many bit strings of 
length 10 do not have three consecutive 1s?

We could use an approach similar to the previous example and produce all bit strings of length 10 
and then count the number that do not contain three consecutive 1s.  However, the solution to 
Example 21 of Section 6.1, and especially Figure 2, suggests a more efficient solution.  Instead of 
creating all the possible bit strings, we can build them in such a way as to only create those that 
satisfy the limitation on the number of consecutive 1s.  

To solve this problem, we will use a recursive algorithm.  The basis step will be the set consisting 
of all bit strings of length 2 (these cannot have three consecutive 1s).  In the recursive step, the new 
version of the set will be constructed as follows.  For each bit string in the previous set, we will 
append a 0.  And we will append a 1 to all of the bit strings whose last two bits are not both 1.  

For this problem, we will model bit strings as lists of 0s and 1s.  The algorithm described above 
will be implemented as two procedures.  The first procedure will be responsible for extending a 
particular bit string.  That is, given a bit string (a list of 0s and 1s), it will return either the two bit 
strings obtained by appending a 0 and by appending a 1, or it will return the single bit string formed
by appending a 0 if appending a 1 would result in three consecutive 1s.  The second procedure will 
apply the first to an entire set of bit strings.

First we implement the procedure that extends a single bit string.  The parameter to this procedure 
will be a single bit string.  The procedure will first create a new bit string by appending a zero to the 
input.  

Then the procedure will test the last two elements of the original bit string to determine if they are 
both ones.  We will accomplish this test by extracting the sublist consisting of the last two entries 
with the selection operator: L[-2..-1].  We can then compare the result against the list [1,1].  
If those lists are equal, that is, the last two bits are both 1, then the procedure only returns the list 
obtained by adding 0.  Otherwise, it will create the list with 1 added and return both extended bit 
strings as a sequence.

Here is the implementation. 
AddBit := proc(L::list)
  local L0, L1;
  L0 := [op(L),0];
  if L[-2..-1]=[1,1] then
    return L0;
  else
    L1 := [op(L),1];
    return L0,L1;
  end if;
end proc:

Let's test this procedure on two examples: 1, 0, 1, 1  should produce only 1, 0, 1, 1, 0 , while 
applying the procedure to that result should produce 1, 0, 1, 1, 0, 0  and 1, 0, 1, 1, 0, 1 .

AddBit([1,0,1,1]);
1, 0, 1, 1, 0

AddBit(%);
1, 0, 1, 1, 0, 0 , 1, 0, 1, 1, 0, 1

Now we write the main procedure.  It will accept as input a positive integer n representing the length
of the bit strings to be output.  In case this value is 2, it will return the four bit strings of length 2.  
For values of n larger than 2, it will recursively call itself on nK 1 and store the result of the 
recursion as S.  It then initializes a new list T to the empty set.  Finally, it loops through the set S, 
applying AddBit to each member and adding the result to T.

Here is the implementation.
FindBitStrings := proc(n::nonnegint)
  local S, s, T;
  if n = 2 then
    return {[0,0],[0,1],[1,0],[1,1]};
  else
    S := FindBitStrings(n-1);
    T := {};
    for s in S do
      T := T union {AddBit(s)};
    end do;
    return T;
  end if;
end proc:

Applying the procedure to 10 and using nops will give us the number of bit strings of length 10 

that do not include three successive 1s.
nops(FindBitStrings(10));

504

6.2 The Pigeonhole Principle
In this section we will see how Maple can be used to help explore two problems related to the 
pigeonhole principle: finding consecutive entries in a sequence with a given sum, and finding 
increasing and decreasing subsequences.

Before considering those two problems, however, recall the ceil command.  This command 
calculates the ceiling of an expression.  For example, the solution to Example 8 in the text indicates 
that the minimum number of area codes needed to assign different phone numbers to 25 million 

phones is 
25000000
8000000

.  In Maple, this can be computed by the following statement.

ceil(25000000/8000000);
4

Consecutive Entries with a Given Sum
Example 10 in Section 6.2 describes the solution to the following problem.  In a month with 30 
days, a baseball team plays at least one game per day but at most 45 games during the month.  There
then must be a period of consecutive days during which the team plays exactly 14 games.

The problem can be generalized.  Given a sequence of d positive integers whose sum is at most S, 
there must be a consecutive subsequence with sum T for any T ! SK d.  We leave it to the reader 
to prove this assertion.

We will write a procedure that, given a sequence and target sum T, will find the consecutive terms 
whose sum is T.  Our solution will be based on the approach used in Example 10.  

First, we will calculate the numbers a1, a2,…, ad with each aj equal to the sum of the first j terms in 
the sequence.  These values will be stored as a list, A.  We will calculate these sums using the 
observation that each one is equal to the previous sum plus the next entry in the sequence.

Then we will calculate ai CT for each i and use the member command to check to see if this value 
is in A.  The member command requires two arguments: the first is the element being sought and 
the second is the list (or set) to be searched.  The command returns true if the element is found and 
false otherwise.  The member command also accepts an unevaluated (i.e., enclosed in right single 
quotes) name as a third, optional argument.  If the element is found in the list, then this name is 
assigned to the position of the first occurrence.  For example,

member("d",["a","b","c","d","e","f","g"],'p');
true

p;
4

Finally, if ai CT is found in the list a1, a2,…, ad, say at position j, then we know that i through j 
are the positions of the consecutive subsequence with the desired sum.  The procedure returns the 
starting and ending positions as well as the subsequence.

FindSubSum := proc(L::list(posint),T::posint)
  local A, i, j, a, p;
  A := [L[1]];
  for i from 2 to nops(L) do
    A := [op(A),A[i-1]+L[i]];
  end do;
  for i from 1 to nops(L) do
    if member(A[i]+T,A,'j') then
      return i,j,L[i..j];
    end if;
  end do;
  return FAIL;
end proc:

We can apply our procedure to the following sequence, representing the number of games a baseball
team played on each day of a 30-day month: 

2, 1, 3, 1, 1, 3, 1, 1, 1, 1, 3, 1, 3, 1, 1, 1, 1, 1, 1, 1, 1, 1, 3, 1, 1, 3, 1, 3, 1, 1.
As in Example 10, we'll find the consecutive days during which the team played 14 games.

FindSubSum([2, 1, 3, 1, 1, 3, 1, 1, 1, 1, 3, 1, 3, 1, 1, 1, 
1, 1, 1, 1, 1, 1, 3, 1, 1, 3, 1, 3, 1, 1],14);

5, 13, 1, 3, 1, 1, 1, 1, 3, 1, 3

Strictly Increasing Subsequences
Theorem 3 of Section 6.2 asserts that every sequence of n2 C 1 distinct real numbers contains a 
subsequence of length nC 1 that is either strictly increasing or strictly decreasing.  We will develop 
a procedure that will find a longest strictly increasing subsequence.  

The Patience algorithm
To find the longest increasing subsequence, we will use a greedy strategy based on "Patience 
sorting" (the name refers to the solitaire card game also called Klondike).  The idea is as follows.  
Imagine that the numbers in the sequence are written on cards.  The cards are placed in a "deck" in 
the order they appear in the sequence and with the first element of the sequence on top.  Now play a 
"game" using the deck of cards based on the following rules.

The cards are "dealt" one at a time onto a series of piles on the table.  Initially there are no piles.  The
top card (the first element in the sequence) is the first card dealt and forms the first pile.  To play the 
next card, check to see if it is less than or greater than the first card.  If the second card (the second 
element of the sequence) is less than the first, then it is placed on the first pile, on top of the first 
card.  If the second card is greater than the first, then it starts a new pile to the right of the first.

The "game" continues in this way.  At each step, the table has on it a series of piles.  To play the 
next card, you compare the value on the card to the card on top of the first pile.  If the card to be 
played has a value smaller than the number showing on the first pile, then the new card is placed on 
top of the first pile.  Otherwise, you look at the second pile.  If the card being played is smaller than 
the value on the second pile, it is placed on top of the second pile.  Continue in this fashion until 
either the card has been played or, if it is larger than the top card on every existing pile, then it 
begins a new pile to the right of all the others.

An illustration is in order.  Consider the sequence 12, 18, 7, 11, 16, 3, 20, 17.



Step 1: play the first entry, 12, as the first pile
12

Step 2: play the second entry, 18.  Since 
18 O 12, 18 starts a new pile. 12 18

Step 3: play 7.  Checking the first pile, note that
7 ! 12, so 7 is played on the first pile. 7

12 18

Step 4: play 11.  Checking the first pile, 
11 O 7, so do not play 11 on first pile.  
Checking the second pile, 11 ! 18, so play 11 
on the second pile.

7 11

12 18

Step 5: play 16.  Checking the first pile, 
16 O 7 so do not play 16 on the first pile.  
Checking the second pile, 16 O 11, so 16 
begins a third pile.

7 11

12 18 16

Step 6: play 3.  Checking the first pile, 3 ! 7, 
so 3 is played on the first pile. 3

7 11

12 18 16

Step 7: play 20.  Checking the first pile, 
20 O 3.  Checking the second pile, 20 O 11.  
Checking the third pile, 20 O 16, so 20 starts a 
new pile.

3

7 11

12 18 16 20

Step 8: play 17.  Checking the first pile, 
17 O 3.  Checking the second, 17 O 11.  
Checking the third, 17 O 16, but 17 ! 20, so 
17 is played on the fourth pile.

3

7 11 17

12 18 16 20

Once the "game" is complete, the length of the longest strictly increasing subsequence is equal to the
number of piles.

Now that all of the cards are played, we obtain a strictly increasing subsequence by backtracking.  
The top card on the final pile is 17, so 17 will be the last entry in the subsequence.  When 17 was 
placed on the pile, the top card on the pile before it was 16 (step 8), so 16 precedes 17 in the 
subsequence.  When 16 was placed on the third pile, the top card on the second pile was 11 (step 5),
so 11 precedes 16.  And when 11 was placed on the second pile, the top card on the first pile was 7,

so 7 is first in the subsequence.  Thus, 7, 11, 16, 17 is a strictly increasing subsequence of maximal 
length.

You are encouraged to "play" through this approach a few times with your own sequences to 
ensure that you understand the process before continuing on to the implementation of the procedure 
below.  You can apply the FindIncreasing procedure defined below to make sure that you are 
arriving at the same result.  Be sure to keep track of what was on top of the previous pile when each
number is played, as you need that information in the backtracking stage.

Implementing the algorithm
Now we will implement the Patience algorithm.  The given sequence will be input to the procedure 
as a list.  Within the procedure, we need to track three kinds of information.  

First, we need to know which "step" we're in, that is, which card is being played.  This will be 
represented by the variable to a for loop ranging from 1 to the size of the list.

Second, we need to know what cards are on the piles.  Specifically, we need to know the top card of
each pile.  This will be represented as a list, piles.  When a card is placed on top of an existing 
pile, we can replace the current value in that position with the the selection-assignment syntax
piles[i] := x;.  When a new pile is added to the list, we extend the list via the usual syntax
[op(piles),x];.

Third, in order to backtrack and recover the longest increasing sequence, we need to store, for each 
member of the sequence, the value that was on the top of the pile to the left of the entry's pile.  For 
this, we will use a table, pointers, whose indices will be the members of the sequence and 
whose entries will be set to the previous pile's top card.  (We use the name pointers for this table
because of the similarity to the "linked list" structure used in some programming languages.)  For 
those numbers played in the first pile, the value in the table will be set to NULL.

The algorithm will consist of two stages.  The first stage will be the game stage.  After initializing
piles to the empty list and pointers to the empty table, we begin a for loop with loop variable
step running from 1 to the length of the input sequence S.  Within this main loop, two tasks are 
performed.  

The first thing that happens within the step loop is determining on which pile to play the current 
card.  Note that the value of the current card is accessed by S[step].  To determine the proper 
location for the current card, we do the following.  

Initialize a variable whichpile to 0.  
Use a for loop from 1 to the current number of piles.  Within the for loop, compare the current 
card to the top of each pile.  If the current card is smaller than the value in piles, set
whichpile equal to that pile index.  
Once the loop exits, check the value of whichpile.  If it is 0 following termination of the 
loop, that means a pile was not found for the current card, and thus the piles list must be 
extended to create a new pile for this card.  Otherwise, the value of whichpile is not 0 and 
we update the corresponding entry in piles to indicate that the latest card is placed on top in 
that position.

The second task within the step loop is to update the pointers table.  This also depends on the 
value of whichpile.  

If whichpile is 1 or if piles only contains 1 entry, then the latest card was played on the 
first pile and the associated value should be NULL. 

• • If whichpile is 0, then the value associated with the latest card is piles[-2], the top card 
on what had been the last pile but is now the next to last pile.  (Note that the previous condition, 
that whichpile is 1 or piles has only 1 entry will ensure that this second condition is only 
tested if piles has at least 2 entries and thus piles[-2] is a valid selection.)  
Otherwise, the value is piles[whichpile-1].

That concludes the game stage.  The second stage is the backtracking stage, which is much simpler. 
First, access the top card of the last pile with piles[-1], and initialize the maximal increasing list,
iList, to the list consisting of this value.  

Then we extend iList on the left with the entry in the pointers table associated to that value.  
Since we're building the list from right to left, iList[1] always contains the most recently added 
number.  So pointers[iList[1]] is the new value, which is added via the expression
[pointers[iList[1]],op(iList)].  Since the cards played in the first pile were associated
with NULL, we can use a while loop with condition pointers[iList[1]] <> NULL to fill the
iList.  At the conclusion of the loop, the procedure returns iList.

Here, finally, is the procedure.
FindIncreasing := proc(S::list)
  local piles, pointers, step, whichpile, p, iList;
  piles := [];
  pointers := table();
  # game stage
  for step from 1 to nops(S) do
    # playing the card
    whichpile := 0;
    for p from 1 to nops(piles) do
      if S[step] < piles[p] then
        whichpile := p;
        break;
      end if;
    end do;
    if whichpile = 0 then
      piles := [op(piles),S[step]];
    else
      piles[whichpile] := S[step];
    end if;
    # update pointers
    if whichpile = 1 or nops(piles) = 1 then
      pointers[S[step]] := NULL;
    elif whichpile = 0 then
      pointers[S[step]] := piles[-2];
    else
      pointers[S[step]] := piles[whichpile-1];
    end if;
  end do;
  # backtracking stage
  iList := [piles[-1]];
  while pointers[iList[1]] <> NULL do
    iList := [pointers[iList[1]],op(iList)];
  end do;
  return iList;
end proc:

Example 12 from the text involved the sequence 8, 11, 9, 1, 4, 6, 12, 10, 5, 7.  
FindIncreasing([8,11,9,1,4,6,12,10,5,7]);

1, 4, 5, 7
This is one of the four sequences given in the text.

Connection to the pigeonhole principle
Recall that Theorem 3 asserted that every sequence of n2 C 1 distinct real numbers contains a 
subsequence of length nC 1 that is either strictly increasing or strictly decreasing.  It may appear 
that the Patience algorithm has no connection to this theorem or to the pigeonhole principle.

However, the Patience algorithm does in fact suggest a proof of Theorem 3 via the pigeonhole 
principle.  When the Patience algorithm is executed on a list of n2 C 1 distinct real numbers, either 
there are at least nC 1 stacks or there are at most n stacks.  If there are nC 1 stacks, then there is a 
strictly increasing subsequence of length nC 1. 

On the other hand, assume that there are at most n stacks.  Take the n2 C 1 values to be the pigeons 
and the stacks the pigeonholes.  When n2 C 1 objects are placed in n boxes, then there is a box 

containing at least 
n2 C 1

n
=

n2

n
C

1
n

= nC
1
n

= nC 1 objects (Theorem 2).  Hence some 

stack has nC 1 values.  But the rules of the game ensure that each stack is a strictly decreasing 
subsequence, since one value is placed on top of another in a stack only when the second value is 
lesser than, and appears in the sequence later than, the lower value.  

In short, either there are nC 1 stacks and hence an increasing sequence of length nC 1 or there is a
stack of size nC 1 and hence a decreasing sequence of length nC 1.

6.3 Permutations and Combinations
The combinat package contains many functions pertaining to counting and generating 
combinatorial structures.  We will be using combinat commands extensively in this section.

Permutations
We begin by looking at commands related to permutations of objects.  

We have seen in previous chapters the use of the exclamation mark for factorial.
6!;

720
The factorial command can be used instead of the exclamation mark if you prefer.  Otherwise 
they are equivalent.  (Factorial does not depend on the combinat package.)

Counting permutations
To compute the number of permutations, Maple provides the numbperm command.  This 
command can be used in a few different ways.

You can give only one argument, an integer.  
numbperm(5);

120
In this case, Maple computes the number of permutations, that is, the number of 5-permutations of a
set with 5 elements.  This, of course, is the same as computing 5!.

Instead of an integer as the only argument, you can instead provide a set or list of objects.  
numbperm({"a","b","c","d","e"});

120
With this argument, numbperm computes the number of permutations of the given set (or list).  
Since the set had five elements, this result is the same as the previous value.

To compute the number of r-permutations of a set with n objects, use numbperm with a second 
argument indicating the value of r.  The number of 4-permutations of a set with 7 distinct objects, 
that is P 7, 4 , is computed by the following command.

numbperm(7,4);
840

Once again, the first argument could be given as a set of a list of objects instead of the number of 
objects.

numbperm(["a","b","c","d","e","f","g"],4);
840

You may wonder what the purpose is of allowing the first argument to numbperm to be a set or 
list rather than requiring it to always be a positive integer.  We will explore this more in Section 6.5, 
but briefly, the reason is that the objects in a list do not have to be different.  By giving the first 
element as a list of not-necessarily distinct objects, numbperm will compute the number of 
arrangements of those objects, with repetition.

Listing permutations
To obtain a list of all permutations, Maple provides the permute command.  The syntax is the 
same as numbperm, but instead of reporting the number of permutations, a list of the permutations 
is provided.

Once again, the first argument can be an integer or a set or a list.  If the first argument is given as an 
integer n, Maple will display the permutations of the set 1, 2,…, n .  For example, the following 
command lists the permutations of the first four integers.

permute(4);
1, 2, 3, 4 , 1, 2, 4, 3 , 1, 3, 2, 4 , 1, 3, 4, 2 , 1, 4, 2, 3 , 1, 4, 3, 2 , 2, 1, 3, 4 ,

2, 1, 4, 3 , 2, 3, 1, 4 , 2, 3, 4, 1 , 2, 4, 1, 3 , 2, 4, 3, 1 , 3, 1, 2, 4 , 3, 1, 4,
2 , 3, 2, 1, 4 , 3, 2, 4, 1 , 3, 4, 1, 2 , 3, 4, 2, 1 , 4, 1, 2, 3 , 4, 1, 3, 2 , 4, 2,
1, 3 , 4, 2, 3, 1 , 4, 3, 1, 2 , 4, 3, 2, 1

Providing a specific list or set as the first argument will produce the permutations of the objects in 
the given set or list.  For example, the permutations of the letters a, b, and c are shown below.

permute({"a","b","c"});
"a", "b", "c" , "a", "c", "b" , "b", "a", "c" , "b", "c", "a" , "c", "a", "b" , "c", "b",

"a"

To obtain the r-permutations instead, you must provide r as the second argument.  The following 
lists the 2-permutations of a set with 4 distinct objects.

permute(4,2);
1, 2 , 1, 3 , 1, 4 , 2, 1 , 2, 3 , 2, 4 , 3, 1 , 3, 2 , 3, 4 , 4, 1 , 4, 2 , 4, 3

Or you can provide your own objects to be r-permuted.
permute({"a","b","c","d","e"},2);

"a", "b" , "a", "c" , "a", "d" , "a", "e" , "b", "a" , "b", "c" , "b", "d" , "b",
"e" , "c", "a" , "c", "b" , "c", "d" , "c", "e" , "d", "a" , "d", "b" , "d", "c" ,
"d", "e" , "e", "a" , "e", "b" , "e", "c" , "e", "d"

Random permutations
Maple also provides a command, randperm, that will produce a randomly chosen permutation.

Once again, the argument to randperm can be an integer or a list or set of objects.  If an integer n 
is given, then randperm returns a randomly selected permutation of the set 1, 2,…, n .

randperm(10);
10, 1, 7, 8, 4, 5, 2, 6, 3, 9

If a list or set is provided as the first argument, then the random permutation is produced using the 
elements of the given set or list.

randperm(["a","b","c","d"]);
"c", "b", "d", "a"

Note that the permutation is selected so that each permutation has the same probability of being 
chosen.  

Unlike the numbperm and permute commands, randperm does not accept a second argument. 
(If one is given, it is ignored.)  In order to produce a random r-permutation, you must use one of the
following approaches.

Suppose you need a random 4-permutation of the set "a", "b", "c", "d", "e", "f" .  Your first 
choice is to apply the randperm command to the set, and then use the selection operator to select 
the first four elements.

randperm({"a","b","c","d","e","f"})[1..4];
"e", "b", "f", "c"

The second option is to first use randcomb (described below) to obtain a random subset of size 4 
and then apply randperm to the randomly chosen subset.  This approach will be demonstrated 
below, after describing the commands relevant to combinations.  Both approaches will produce a 
random r-permutation of the given objects.  The second approach is faster to execute, however.

Combinations
The commands related to combinations are very similar to those for permutations.

Counting combinations
The number of combinations is obtained with the numbcomb command.  Like numbperm, this 
command accepts one or two arguments and the first argument can be a number or a set or list.

If you provide only one argument to numbcomb, it returns the total number of combinations. For 
instance,

numbcomb(5);
32

indicates that there are 32 ways to choose some (including all or none) of five objects.  In other 

words, there are 32 subsets of a set of 5 elements.

The first argument can also be a set or a list.
numbcomb({"a","b","c","d","e"});

32

To compute the number of r-combinations, you provide r as the second argument.  Once again, the 
first argument can be a set or list or a positive integer indicating the size of the set the objects are to 
be drawn from.  The following compute C 52, 5  and the number of ways to choose 3 from the set
of vowels.

numbcomb(52,5);
2598960

numbcomb({"a","e","i","o","u"},3);
10

The text mentions that C n, r  is also referred to as a binomial coefficient.  In Maple, the command
binomial can also be used to compute C n, r .  For example, C 52, 5  can be computed by:

binomial(52,5);
2598960

Note that binomial and numbcomb agree when given two nonnegative integers as arguments.  
However, they are not identical commands.  For one, binomial requires two arguments which 
must evaluate to algebraic expressions and cannot accept a list or set as the first argument.  On the 
other hand, binomial uses a more general formula that can compute with rational and floating-
point arguments.  Also, binomial is not part of the combinat package.  

Listing Combinations
The choose command is the combination analog of permute.  

Given only one argument, choose produces all possible combinations of every size.  If the 
argument is a set or list, it will list the subsets or sublists of the argument.  

choose({"a","b","c","d"});
, "a" , "b" , "c" , "d" , "a", "b" , "a", "c" , "a", "d" , "b", "c" , "b",
"d" , "c", "d" , "a", "b", "c" , "a", "b", "d" , "a", "c", "d" , "b", "c", "d" ,

"a", "b", "c", "d"
choose(["a","b","c","d"]);
, "a" , "b" , "c" , "d" , "a", "b" , "a", "c" , "a", "d" , "b", "c" , "b", "d" ,

"c", "d" , "a", "b", "c" , "a", "b", "d" , "a", "c", "d" , "b", "c", "d" , "a", "b",
"c", "d"

Note that if the argument is a set, the result is a set of sets.  On the other hand, if the argument is a 
list, the result is a list of lists.  Many Maple commands accept either sets or lists as arguments, and 
they typically return the same kind of object they were given.  This was not the case for permute, 
because in that case order is relevant, so it must return lists.

If the argument is a non-negative integer n, it will use 1, 2,…, n  by default.
choose(3);

, 1 , 2 , 3 , 1, 2 , 1, 3 , 2, 3 , 1, 2, 3

Note that with only one argument, choose behaves very similarly to powerset, which was 
discussed in Section 2.1 of this manual.  The powerset command accepts only one argument, 
which can be a set, list, or non-negative integer.  It returns the set of all subsets (or the list of all 
sublists) just as choose does.

powerset({"a","b","c"});
, "a" , "b" , "c" , "a", "b" , "a", "c" , "b", "c" , "a", "b", "c"

With a second argument given to choose, you can specify the number of objects to be selected.
choose({"a","b","c","d"},2);

"a", "b" , "a", "c" , "a", "d" , "b", "c" , "b", "d" , "c", "d"
choose(5,2);

1, 2 , 1, 3 , 1, 4 , 1, 5 , 2, 3 , 2, 4 , 2, 5 , 3, 4 , 3, 5 , 4, 5

The powerset command does not allow a second argument.

Random combinations
The randcomb command is used to produce a random combination.  

As opposed to randperm, randcomb requires two arguments.  The first can be a set or list of 
objects or a positive integer, and the second must be a non-negative integer.

randcomb({"a","b","c","d","e"},3);
"a", "c", "d"

randcomb(5,3);
1, 3, 4

As mentioned earlier, randcomb can be combined with randperm to produce a random 
permutation of a specified size.  To obtain a random 4-permutation of "a", "b", "c", "d", "e", "f" , 
for example, first use randcomb to pick a random 4-combination.  Then apply randperm to the 
selected combination.

randcomb({"a","b","c","d","e","f"},4);
"a", "b", "e", "f"

randperm(%);
"f", "e", "b", "a"

You can combine them into one statement, as follows.
randperm(randcomb({"a","b","c","d","e","f"},4));

"a", "e", "b", "c"

Circular Permutations
The prelude to Exercises 40 and 41 of Section 6.3 describes circular permutations.  A circular r-
permutation of n people is a seating of r of those n people at a circular table.  Moreover, two 
seatings are considered the same if one can be obtained from the other by rotation.

The exercises ask you to compute the number of circular 3-permutations of 5 people and to arrive at 
a formula for that number.  In this subsection, we'll write a procedure to list all of the circular r-
permutations of n people.  Having such a procedure can help you more easily explore the concept 
and test your formula.

Rotating a permutation
The key to listing all circular permutations is to devise a way to test whether two circular 
permutations are equal.  According to the definition, two circular permutations are considered to be 
equal if one can be obtained from the other by a rotation.  If we use a list to represent a permutation, 
a rotation will consist of moving the first element to the end of the list (or the last to the front).

Here is a procedure that, given a list representing a circular permutation, will rotate the permutation 
by one position.

RotatePerm := proc(P::list)
  return [op(2..-1,P),P[1]];
end proc:

For example, the seating Abe, Carol, Barbara, is the same as Carol, Barbara, Abe.
RotatePerm(["Abe","Carol","Barbara"]);

"Carol", "Barbara", "Abe"

Note that rotations start to repeat.
RotatePerm(%);

"Barbara", "Abe", "Carol"
RotatePerm(%);

"Abe", "Carol", "Barbara"
In fact, for an r-permutation, after r rotations, the list will return to its original state.

Equality of circular permutations
This observation indicates that, given two r-permutations, we can test to see if they are the same by 
rotating one of them r K 1 times.  The procedure below returns true if the two input lists represent 
the same circular permutation and false otherwise.  It first checks equality without performing 
rotation.  Then, using a for loop, it rotates the second list, checking for equality after each rotation.  

CPEquals := proc(L1::list,L2::list)
  local i, Lr;
  if L1 = L2 then 
    return true;
  end if;
  Lr := L2;
  for i from 1 to nops(L2)-1 do
    Lr := RotatePerm(Lr);
    if L1 = Lr then
      return true;
    end if;
  end do;
  return false;
end proc:

We can use it to confirm equality of circular permutations.  For example, 
CPEquals(["Charles","Helen","Dean"],["Helen","Dean",
"Charles"]);

true

Listing all circular permutations
Now we are prepared to write a procedure that lists all circular permutations.  First, we'll use the
permute command to generate all r-permutations of n people.  We will initialize the set of all 

distinct circular permutations to the the first element of the list of all permutations.  That first 
permutation is then removed from the list of all permutations.

Within a while loop, consider the first element in the list of all permutations.  Use CPEquals and a 
loop to see if the first element is identical to any of the members of the set of circular permutations.  
If not, add it to the set of circular permutations.  In either case, it is deleted from the list of all 
permutations.  This continues until the list of all permutations has been emptied.

Here is the implementation.  Our procedure will accept a set as the first argument and r, the number 
that can be seated at the table, as the second argument.

AllCP := proc(S::set,r::posint)
  local allP, allCP, isnew, p;
  allP := combinat[permute](S,r);
  allCP := {allP[1]};
  allP := allP[2..-1];
  while allP <> [] do
    isnew := true;
    for p in allCP do
      if CPEquals(allP[1],p) then
        isnew := false;
        break;
      end if;
    end do;
    if isnew then
      allCP := allCP union {allP[1]};
    end if;
    allP := allP[2..-1];
  end do;
  return allCP;
end proc:

Note that the isnew boolean is used to track whether or not the current first member of allP is 
new or not.  

The following computes the possible circular 3-permutations of the set {"Abe", "Barbara", "Carol", 
"Dean", "Eve"}.

AllCP({"Abe","Barbara","Carol","Dean","Eve"},3);
"Abe", "Barbara", "Carol" , "Abe", "Barbara", "Dean" , "Abe", "Barbara", "Eve" ,

"Abe", "Carol", "Barbara" , "Abe", "Carol", "Dean" , "Abe", "Carol", "Eve" ,
"Abe", "Dean", "Barbara" , "Abe", "Dean", "Carol" , "Abe", "Dean", "Eve" ,
"Abe", "Eve", "Barbara" , "Abe", "Eve", "Carol" , "Abe", "Eve", "Dean" ,
"Barbara", "Carol", "Dean" , "Barbara", "Carol", "Eve" , "Barbara", "Dean",

"Carol" , "Barbara", "Dean", "Eve" , "Barbara", "Eve", "Carol" , "Barbara",
"Eve", "Dean" , "Carol", "Dean", "Eve" , "Carol", "Eve", "Dean"
nops(%);

20

It is left to the reader to experiment with other starting sets and values of r to determine a formula.  
Note that the procedures in this subsection were written using a very naive approach.  There are 
simpler and more efficient approaches, but those would give away the key idea used to create the 
formula.

6.4 Binomial Coefficients and Identities
In this section we will use Maple to compute binomial coefficients, to generate Pascal's Triangle, 
and to verify identities.

The Binomial Theorem

Recall from the previous section that the Maple command binomial can be used to compute 
n
r ,

which is another notation for C n, r .  With n and r positive integers, this command produces the 
same result as numbcomb.  The binomial command is designed to be more general, in that it will
compute coefficients that appear in Newton's generalized binomial theorem.  The generalization is 
beyond the scope of this manual.

Here, we will consider questions such as Examples 2 through 4 from section 6.4 of the text.  

First consider the problem of expanding xC y 5.  In Maple, this can be done easily with the
expand command.  The expand command requires one argument, an algebraic expressions.  It 
returns the result of "expanding" the expression, that is, of distributing products over sums.  

expand((x+y)^5);
x5 C 5 x4 yC 10 x3 y2 C 10 x2 y3 C 5 x y4 C y5

Now consider the question of finding the coefficient of x18y12 in the expansion of xC y 30.  The 

binomial theorem tells us that this coefficient is 
30
12 .  The binomial command with first 

argument 30 and second 12 will produce this value.
binomial(30,12);

86493225

Thus, the expansion of xC y 30 contains the term 86493225 x18y12.

Finding the coefficient of x12y13 in the expansion of 2 xK 3 y 25 requires that we include the 
coefficients of x and y in the computation.  As explained in the solution to Example 4 of the text, the 
expansion is

2 xC K3 y 25 = >
j = 0

25
25
j 2 x 25 K j K3 y j.

The coefficient of x12y13 is found by taking j = 13:
25
13 212 K3 13.

This is
binomial(25,13)*2^12*(-3)^13;

K33959763545702400

Pascal's Triangle
As we've seen, it is very easy to compute binomial coefficients with Maple.  To compute row n of 
Pascal's triangle, we apply the binomial command with the second argument ranging from 0 to n.
This can be done with the seq command.  For example, the 25th row of Pascal's triangle is shown 
below.

seq(binomial(25,k),k=0..25);

1, 25, 300, 2300, 12650, 53130, 177100, 480700, 1081575, 2042975, 3268760, 4457400,
5200300, 5200300, 4457400, 3268760, 2042975, 1081575, 480700, 177100, 53130,
12650, 2300, 300, 25, 1

When calculating a single binomial coefficient or an isolated row, applying the formula may be the 
most efficient approach.  But if you wish to build a sizable portion of Pascal's Triangle, making use 
of Pascal's Identity (Theorem 2 of Section 6.4) and the symmetry property (Corollary 2 of Section 
6.3) can be more effective.

In this subsection, we'll write two procedures for computing binomial coefficients.  The first will 

simply apply the formula 
n!

r ! nK r !
.  The second will be a recursive procedure making use of 

Pascal's Identity and symmetry.  Then we'll compare the performance of the two procedures in 
building Pascal's Triangle.

The first procedure will be a straightforward application of the formula.  We name it BinomialF 
(for formula).

BinomialF := proc(n::nonnegint, k::nonnegint)
  return n!/(k!*(n-k)!);
end proc:

A recursive procedure
The second procedure will be called BinomialR (for recursive).  Recall Pascal's Identity:

nC 1
k =

n
kK 1 C

n
k .

Rewriting this in terms of n and nK 1, we have
n
k =

nK 1
kK 1 C

nK 1
k .

Also recall that the binomial coefficients are symmetric, that is,
n
k =

n
nK k .

With these facts in mind, our recursive procedure will work as follows.  We will declare the
remember option so that Maple will automatically create a remember table.  The body of the 
procedure will consist of an if-elif-else statement.  In case the second argument, is 0, the procedure 
will return 1.  That forms the basis case of the recursion.  The elif clause will test whether 2 k O n.  
In this case, we make use of symmetry and call BinomialR on n and nK k.  Finally, in the else 
clause, we apply Pascal's Identity, making recursive calls to BinomialR.  Here is the 
implementation.

BinomialR := proc(n::nonnegint, k::nonnegint)
  option remember;
  if k=0 then
    return 1;
  elif 2*k > n then
    return BinomialR(n,n-k);
  else
    return BinomialR(n-1,k-1) + BinomialR(n-1,k);
  end if;
end proc:

We use our two procedures to build Pascal's triangle using a loop to control the value of n and the
seq command.

for n from 0 to 5 do
  seq(BinomialF(n,k),k=0..n);
end do;

1
1, 1

1, 2, 1
1, 3, 3, 1

1, 4, 6, 4, 1
1, 5, 10, 10, 5, 1

for n from 0 to 5 do
  seq(BinomialR(n,k),k=0..n);
end do;

1
1, 1

1, 2, 1
1, 3, 3, 1

1, 4, 6, 4, 1
1, 5, 10, 10, 5, 1

Comparing performance
Now we'll compare the performance of the two procedures.  

First, we'll use each of them to compute the first 1000 rows of Pascal's Triangle and compare the 
time it takes.  We must use forget to clear the remember table for BinomialR.

forget(BinomialR);
st := time():
for n from 0 to 1000 do
  seq(BinomialF(n,k),k=0..n):
end do:
time() - st;

20.502
st := time():
for n from 0 to 1000 do
  seq(BinomialR(n,k),k=0..n):
end do:
time() - st;

2.253
You see that the difference between the two performance of the two procedures is substantial.

However, if we compute some isolated values, the situation is reversed.  We will generate some 
random values of n between 100 and 1000 and random values of k between 0 and 100.  

Recall that the rand command applied to a range of integers returns a procedure that generates 
random numbers in that range.  For instance, the statement below assigns to the name randN a 
procedure.  Calling randN() produces a random number.

randN := rand(100..1000):
randN();

296
Likewise for randK.  Note that calling rand with a single integer as the argument is the same as 
the range from 0 to that value.

randK := rand(100):
randK();

26

Now we'll generate a list of 100 randomly generated n, k  pairs.  We will suppress the output since
the list will be long.  But note that the elements of this list are 2-element lists.

NKlist := [seq([randN(),randK()],i=1..100)]:

Now we apply the two procedures to the elements of the list.  Note that we pass the arguments to 
both procedures with the syntax BinomialF(op(NKpair)).  The op command applied to
NKpair, an element of the NKlist, takes the 2-element list NKpair and returns the underlying 
expression sequence.  The procedures were written to accept two integer arguments, so passing 
them the 2-element list (which is only one argument) would raise an error.

st := time():
for NKpair in NKlist do
  BinomialF(op(NKpair)):
end do:
time() - st;

0.004
forget(BinomialR);
st := time():
for NKpair in NKlist do
  BinomialR(op(NKpair)):
end do:
time() - st;
  

0.276

The reason for the difference in relative performance is that when generating Pascal's Triangle, all of
the values beginning with n = 0, k = 0 needed to be calculated.  On the other hand when calculating 
certain isolated values, the recursive procedure still had to calculate all of the results for lower values
of n and k, which the non-recursive procedure did not need to compute.

Verifying Identities
Maple can help verify identities regarding the binomial coefficients.  If you enter an expression 
using the binomial command that includes symbolic arguments, Maple will echo the statement.  
First, we ensure that n and k are unassigned.

n := 'n'; k := 'k';
n := n
k := k

binomial(n,k);
binomial n, k

To have Maple produce an expression using factorials, use the convert command with the 

binomial expression as the first argument and the keyword factorial as the second argument.  
convert(binomial(n,k),factorial);

n!
k! nK k !

Verifying symmetry
As a first example, we'll verify the identity C n, k = C n, nK k , the symmetry identity.

Assign names to the left and right hand sides of the identity.
left := binomial(n,k);

left := binomial n, k
right := binomial(n,n-k);

right := binomial n, nK k

Remember that we need to use the evalb command to evaluate boolean expressions, such as the 
equality left=right.

evalb(left=right);
false

The negative result illustrates that a great deal of care is needed when using Maple to verify 
identities.  Maple does not recognize the above as a true statement and therefore reports false.  That 
does not mean that the identity is not in fact true.  It only indicates that Maple was not able to 
confirm it.

In order to verify the symmetry identity, we need to apply two other commands.  First, we'll use the
convert command to replace the definitions of left and right.

left := convert(left,factorial);

left :=
n!

k! nK k !
right := convert(right,factorial);

right :=
n!

k! nK k !

Second, we'll use simplify to help ensure that Maple will perform necessary algebraic 
manipulations in order for it to recognize the equality.  It is often simplest to embed the call to
simplify within the evalb statement.  (In this example, simplify is not necessary, but it is 
typically needed.)

evalb(simplify(left=right));
true

The hockeystick identity
Exercise 27 in Section 6.4 asks you to prove the hockeystick identity:

>
k = 0

r
nC k

k =
nC r C 1

r .

The exercise asks you to prove it using a combinatorial argument and using Pascal's identity.  Maple
will verify this identity for us using algebra based on the formula for the binomial coefficient.

First, we will give the right hand side of the identity a name.
rightHockey := binomial(n+r+1,r);

rightHockey := binomial nC r C 1, r

The left hand side is a summation.  Since it is a symbolic sum, we must use sum, rather than add 
to represent it.

leftHockey := sum(binomial(n+k,k),k=0..r);

leftHockey :=
r C 1  binomial nC r C 1, r C 1

nC 1
Notice that Maple was able to use information it knows about binomial coefficients in order to do 
some simplification of the summation.  

To verify the identity, we need to go a step further and convert the expressions to factorials.
leftHockey := convert(leftHockey,factorial);

leftHockey :=
r C 1  nC r C 1 !
nC 1  r C 1 ! n!

rightHockey := convert(rightHockey,factorial);

rightHockey :=
nC r C 1 !
r ! nC 1 !

Now evalb combined with simplify will confirm the identity.
evalb(simplify(leftHockey=rightHockey));

true

Keep in mind when using Maple to check identities that it will only report true when the two sides 
of the expression are identical.  If it does report true, you can be fairly confident that the identity 
does hold, though a truly convincing proof requires that you explicitly show the algebraic 
manipulations.  

If Maple reports false, however, even after using convert and simplify, you can not be certain
whether the identity is false or if it is true but more manipulation is needed to get Maple to recognize
it.  To use Maple to demonstrate that a purported identity is false, you would find a counterexample 
by computing the values of both expressions and finding inputs that result in different values.  
(Refer to Section 1.7 for examples of finding counterexamples.)

6.5 Generalized Permutations and Combinations
In this section we will introduce a variety of Maple commands related to permutations and 
combinations with repetition allowed and related to distributing objects in boxes where the objects 
and the boxes may or may not be distinguishable.  

Permutations with Repetition
Recall from Theorem 1 that the number of r-permutations of n objects is nr if repetition is allowed.

For example, the number of strings of length 5 that can be formed from the 26 uppercase letters of 
the English alphabet is

26^5;
11881376

As a second example, we compute the number of ways that four elements can be selected in order 

from a set with three elements when repetition is allowed.
3^4;

81

Recall from the previous section that the numbperm and permute commands can accept either a 
number or a list or a set as the first argument.  In case you provide a list as the first argument, that 
list may have repeated elements.  In case the list does contain repeated elements, those are allowed to
repeat in the permutations.

For example, consider the statement below.
permute([1,1,2]);

1, 1, 2 , 1, 2, 1 , 2, 1, 1
Given a list of n not necessarily distinct objects and no second argument, the permute command 
produces all of the n-permutations of the n objects.  Note that 1 appeared twice in the input and thus 
appears twice in the results, while 2 appeared once in the input and so appears once in the output.

With a second argument, you can specify the length of the permutations.
permute([1,1,2],2);

1, 1 , 1, 2 , 2, 1
Since 1 appeared twice in the input list, it was allowed to appear twice in the results, but 2 appeared 
only one time in the input and thus was not allowed to repeat.

This means that if you want to list the ways that four elements can be selected in order from a set 
with three elements when repetition is allowed, you can use the permute command, so long as 
you repeat the elements in the input.  In order to generate all r-permutations of n objects with 
repetition allowed, you must use as input the list consisting of the n objects each repeated r times.  If
an object is repeated fewer than r times in the input list, then that will limit the number of times it is 
allowed to repeat in the results.

To form 4-permutations with repetition allowed of "a", "b", "c" , we apply the permute 
command as shown below.  Recall the use of the dollar sign operator with syntax o $ n to 
produce a sequence of the object o repeated n times.

abcRepeated := permute(["a" $ 4, "b" $ 4, "c" $ 4],4);
abcRepeated := "a", "a", "a", "a" , "a", "a", "a", "b" , "a", "a", "a", "c" , "a", "a",

"b", "a" , "a", "a", "b", "b" , "a", "a", "b", "c" , "a", "a", "c", "a" , "a", "a", "c",
"b" , "a", "a", "c", "c" , "a", "b", "a", "a" , "a", "b", "a", "b" , "a", "b", "a", "c" ,
"a", "b", "b", "a" , "a", "b", "b", "b" , "a", "b", "b", "c" , "a", "b", "c", "a" ,
"a", "b", "c", "b" , "a", "b", "c", "c" , "a", "c", "a", "a" , "a", "c", "a", "b" , "a",

"c", "a", "c" , "a", "c", "b", "a" , "a", "c", "b", "b" , "a", "c", "b", "c" , "a", "c",
"c", "a" , "a", "c", "c", "b" , "a", "c", "c", "c" , "b", "a", "a", "a" , "b", "a", "a",
"b" , "b", "a", "a", "c" , "b", "a", "b", "a" , "b", "a", "b", "b" , "b", "a", "b",
"c" , "b", "a", "c", "a" , "b", "a", "c", "b" , "b", "a", "c", "c" , "b", "b", "a", "a" ,
"b", "b", "a", "b" , "b", "b", "a", "c" , "b", "b", "b", "a" , "b", "b", "b", "b" ,
"b", "b", "b", "c" , "b", "b", "c", "a" , "b", "b", "c", "b" , "b", "b", "c", "c" ,
"b", "c", "a", "a" , "b", "c", "a", "b" , "b", "c", "a", "c" , "b", "c", "b", "a" ,

"b", "c", "b", "b" , "b", "c", "b", "c" , "b", "c", "c", "a" , "b", "c", "c", "b" ,
"b", "c", "c", "c" , "c", "a", "a", "a" , "c", "a", "a", "b" , "c", "a", "a", "c" , "c",

"a", "b", "a" , "c", "a", "b", "b" , "c", "a", "b", "c" , "c", "a", "c", "a" , "c", "a",
"c", "b" , "c", "a", "c", "c" , "c", "b", "a", "a" , "c", "b", "a", "b" , "c", "b", "a",
"c" , "c", "b", "b", "a" , "c", "b", "b", "b" , "c", "b", "b", "c" , "c", "b", "c",
"a" , "c", "b", "c", "b" , "c", "b", "c", "c" , "c", "c", "a", "a" , "c", "c", "a", "b" ,
"c", "c", "a", "c" , "c", "c", "b", "a" , "c", "c", "b", "b" , "c", "c", "b", "c" , "c",

"c", "c", "a" , "c", "c", "c", "b" , "c", "c", "c", "c"
nops(abcRepeated);

81

Note that the size of the list produced by permute agrees with the answer given by the formula nr.

The numbperm command has the same syntax as the permute command and will return the 
number of permutations without listing them all.

numbperm(["a" $ 4, "b" $ 4, "c" $ 4],4);
81

Combinations with Repetition
Combinations with repetition can be handled in Maple in much the same way as permutations with 
repetition are.

Theorem 2 of Section 6.5, asserts that the number of r-combinations of a set of n objects when 
repetition of elements is allowed is C nC r K 1, r .  This suggests the following useful functional 
operator.

numbcombrep := (n,r) -> numbcomb(n+r-1,r):
Thus we can compute, for example, the number of ways to select five bills from a cash box with 
seven types of bills (Example 3) as follows.

numbcombrep(7,5);
462

We can also make use of the numbcomb and choose commands in the same way as numbperm 
and permute were used above.  

Consider Example 2 from Section 6.5.  In this example, we are given a bowl of apples, oranges, and
pears and are to select four pieces of fruit from the bowl provided that it contains at least four pieces
of each kind of fruit.  We have three ways to solve this problem with Maple.

First, we can use the numbcombrep functional operator that we created.
numbcombrep(3,4);

15

The second approach is to use numbcomb.  The first argument is the list of "apple", "orange", and 
"pear" each repeated four times, and the second argument is 4 indicating that four objects are to be 
selected.

numbcomb(["apple" $ 4, "orange" $ 4, "pear" $ 4],4);
15

The third option is to use choose to list all the options.  The arguments are the same as they were 
for numbcomb.

choose(["apple" $ 4, "orange" $ 4, "pear" $ 4],4);
"apple", "apple", "apple", "apple" , "apple", "apple", "apple", "orange" , "apple",

"apple", "apple", "pear" , "apple", "apple", "orange", "orange" , "apple", "apple",
"orange", "pear" , "apple", "apple", "pear", "pear" , "apple", "orange", "orange",
"orange" , "apple", "orange", "orange", "pear" , "apple", "orange", "pear", "pear" ,
"apple", "pear", "pear", "pear" , "orange", "orange", "orange", "orange" ,
"orange", "orange", "orange", "pear" , "orange", "orange", "pear", "pear" ,
"orange", "pear", "pear", "pear" , "pear", "pear", "pear", "pear"
nops(%);

15

Note that when repetition is allowed, all of the commands numbperm, permute, numbcomb, and
choose must be given a list as the first argument.  If you were to use a set instead of a list, the 
repeated elements in the set would be automatically removed by Maple.  

Permutations with Indistinguishable Objects
Maple handles permutations with indistinguishable objects in the same way as when repetition is 
allowed.  The numbperm and permute commands both accept lists of objects as their first 
argument.  If objects in the list are repeated, Maple considers them as indistinguishable and counts 
the permutations appropriately.

For example, to solve Example 7, finding the number of different strings that can be made from the 
letters of the word SUCCESS, we use the list S, U, C, C, E, S, S  as the argument to numbperm.

numbperm(["S","U","C","C","E","S","S"]);
420

Observe that this gives the same result as the formula given in Theorem 3.  

Maple makes it easy to go a bit further than Theorem 3, which is restricted to the situation when you
are permuting all n objects.  To find the number of r-permutations of n objects where some objects 
are indistinguishable, you give r as the second argument.

For example, the number of strings of length 3 that can be made from the letters of the word 
SUCCESS can be calculated as follows.

numbperm(["S","U","C","C","E","S","S"],3);
43

Listing these words can be accomplished by use of the permute command with the same 
arguments.

permute(["S","U","C","C","E","S","S"],3);
"S", "U", "C" , "S", "U", "E" , "S", "U", "S" , "S", "C", "U" , "S", "C", "C" ,

"S", "C", "E" , "S", "C", "S" , "S", "E", "U" , "S", "E", "C" , "S", "E", "S" ,
"S", "S", "U" , "S", "S", "C" , "S", "S", "E" , "S", "S", "S" , "U", "S", "C" ,
"U", "S", "E" , "U", "S", "S" , "U", "C", "S" , "U", "C", "C" , "U", "C",

"E" , "U", "E", "S" , "U", "E", "C" , "C", "S", "U" , "C", "S", "C" , "C",

"S", "E" , "C", "S", "S" , "C", "U", "S" , "C", "U", "C" , "C", "U", "E" ,
"C", "C", "S" , "C", "C", "U" , "C", "C", "E" , "C", "E", "S" , "C", "E",

"U" , "C", "E", "C" , "E", "S", "U" , "E", "S", "C" , "E", "S", "S" , "E", "U",
"S" , "E", "U", "C" , "E", "C", "S" , "E", "C", "U" , "E", "C", "C"

A related question is the number of strings with 3 or more letters that can be made from the letters 
of the word SUCCESS.  To do this, we compute the number of r-permutations with r equal to 3, 4, 
5, 6, and 7 and sum these values.  Using the add command, this can be done in one statement.  

Remember that the add command accepts a first argument in terms of a variable such as i.  With 
the second argument specifying the range of values that i can take, the command returns the sum of
the numbers obtained by evaluating the first argument at each value for i.  

add(numbperm(["S","U","C","C","E","S","S"],i),i=3..7);
1247

Distinguishable Objects and Distinguishable Boxes
Example 8 asks how many ways there are to distribute hands of 5 cards to each of four players 
from a deck of 52 cards.  There are several ways to compute this value in Maple.

First, we can use the expression in terms of combinations, C 52, 5 C 47, 5 C 42, 5 C 37, 5  by
using numbcomb.

numbcomb(52,5)*numbcomb(47,5)*numbcomb(42,5)*numbcomb(37,5);
1478262843475644020034240

Second, we can use the formula from Theorem 4: 
52!

5!5!5!5!32!
.

52!/(5!*5!*5!*5!*32!);
1478262843475644020034240

Note that when multiplying factorials, the asterisk is required.

Finally, this same value can be computed using the multinomial command.  This command 
requires at least two positive integers as arguments, but can accept any number of positive integers.  
Regardless of the number of arguments, the first must be equal to the sum of the rest.  If n is the 
first argument and k1, k2,…, km are the rest of the arguments, then assuming k1 C k2 C/C km = n, 
the command returns

n!
k1 !k2 !/km!

.

That is, multinomial implements the formula of Theorem 4.  (Exercise 63 of Section 6.5 
explains the reason for the term multinomial.)

Computing the number of ways to deal cards, as described above, can be computed as follows.
multinomial(52,5,5,5,5,32);

1478262843475644020034240

Revising the multinomial command
It is common in questions about distributing distinguishable objects into distinguishable boxes that 
you want to distribute only some of the objects.  In Example 8, for instance, not all of the cards are 



(6.101)(6.101)

O O 

O O 
(6.106)(6.106)

O O 

(6.64)(6.64)

O O 

O O 

(6.107)(6.107)

(6.91)(6.91)

• • 

distributed.  The multinomial command requires that you include the remainder of the cards as 
an argument.  Conceptually, you can think of making one more box to hold the objects that are not 
placed in any of the other boxes.  

This is such a common occurrence, however, that it seems more natural to forget about this "discard
box."  We will write a Maple command that will use the formula from Theorem 4 but will not 
require that we provide the size of the discard box.  

In order to make our procedure work as much like multinomial as possible, we need it to be 
able to accept any number of arguments.  

To do this, we apply the seq modifier to the parameter.  As a simple example of how this works, 
consider the example below.

printInts := proc(x::seq(integer))
  print(x);
end proc:

The parameter declaration x::seq(integer) indicates that the parameter is to accept a sequence 
of integers.  When the procedure is called with a sequence of integers as arguments, the name x is 
assigned to the sequence.

printInts(5,2,9,11);
5, 2, 9, 11

If non-integers are included in the call to the procedure, the parameter will only be assigned to the 
sequence of integers up until the first non-integer.

printInts(-5,8,2,"hello",7,11);
K5, 8, 2

The type in parentheses after the seq keyword can be any Maple type.  Also note that such a 
parameter can match the empty sequence in which case it is assigned NULL.

Our procedure will have two parameters.  The first will be a positive integer n.  The second will be 
the sequence of positive integers k.  First, we ensure that k is not empty.  (If k = NULL, the 
procedure returns 1, as there is 1 way to discard all the objects.)  Then we calculate the difference of
the parameter n and the sum of the rest of the arguments.  Note that k must be turned into a list to be
used with the add (and most other) commands.  The difference is the number of objects that are 
discarded.  If this is negative, the procedure will raise an error.  Otherwise, it becomes the final km in
the Theorem 4 formula.

Here is the implementation.  Note that we use the map command to compute the factorials of each of
the integers in the denominator of the formula.  

distinguishable := proc(n::posint, k::seq(posint))
  local discard, denomList, i;
  if k = NULL then
    return 1;
  end if;
  discard := n - add(i,i=[k]);
  if discard < 0 then
    error "n must be at least the sum of the rest."
  end if;
  denomList := [k,discard];
  denomList := map(factorial,denomList);



O O 

(6.101)(6.101)

O O 

(6.109)(6.109)
O O 

(6.108)(6.108)

(6.110)(6.110)

O O 

(6.64)(6.64)

O O 

O O 

(6.91)(6.91)

• • 

  return n!/mul(i,i=denomList);
end proc:

With this procedure, we can compute the number of ways to deal five cards to each of four players 
from a deck of 52 cards as follows.

distinguishable(52,5,5,5,5);
1478262843475644020034240

Indistinguishable Objects and Distinguishable Boxes
The text describes the correspondence between questions about placing indistinguishable objects 
into distinguishable boxes and about combination with repetition questions.  

Example 9 asks how many ways 10 indistinguishable balls can be placed in 8 bins.  We can use the
numbcombrep function written earlier.

numbcombrep(8,10);
19448

It may seem that the arguments were reversed.  Keep in mind that the connection to combinations 
with repetition is that you are selecting 10 bins from the 8 available bins with repetition allowed.

Compositions and weak compositions
We can also use the composition and numbcomp commands to answer questions of this kind.  
A k-composition of a positive integer n is a way of writing n as the sum of k positive integers where
the order of the summands matters.  For example, 4 has three distinct 2-compositions: 3C 1, 2C 2,
and 1C 3.  

A weak composition is similar, but the terms in the sum are allowed to be 0.  Thus 4 has five 
distinct weak 2-compositions: 4C 0 and 0C 4 in addition to the three listed before. 

Note that the weak r-compositions of n correspond to the r-compositions of nC r.  For suppose 
that x1 C x2 C/C xr = n is a weak r-composition.  Then each xi is nonnegative.  So 

x1 C 1  C x2 C 1 C/C xr C 1 = nC r,
and each xi C 1 is positive, and hence this is a composition of nC r.  Likewise, any r-composition 
of nC r can be transformed into a weak r-composition of n by subtracting 1 from each term.

Also note that weak r-compositions of n correspond to placing n indistinguishable balls into r 
distinguishable bins.  Suppose x1 C x2 C/C xr = n is a weak r-composition of n.  This can be 
identified with placing x1 of the objects into the first bin, x2 objects in the second bin, etc. 

Counting weak compositions
We now return to numbcomp and composition.  The numbcomp command accepts two 
arguments, n and r.  It returns the number of r-compositions of n.  For example, as we saw, there 
are 3 2-compositions of 4.

numbcomp(4,2);
3

We can create a functional operator to count the number of weak r-compositions of n using the fact 
that this is the same as the number of r-compositions of nC r.

numbweakcomp := (n,r) -> combinat[numbcomp](n+r,r):
We saw there were 5 weak 2-compositions of 4.

numbweakcomp(4,2);
5

And the number of ways that 10 indistinguishable balls can be placed in 8 bins is:
numbweakcomp(10,8);

19448

Also note that weak r-compositions were the subject of Example 5, which asked how many 
solutions there are to x1 C x2 C x3 = 11 with nonnegative integers.

numbweakcomp(11,3);
78

Listing weak compositions
The composition command is used to create a list of all compositions.  The arguments are the 
same as numbcomp.  For example, to list the 2-compositions of 4, enter the following.

composition(4,2);
1, 3 , 2, 2 , 3, 1

We can use composition to create a weakcomposition procedure.  Given arguments n and r,
we apply composition to nC r and r.  For each list in the result, we subtract 1 in each position.

The subtraction of 1 in each position can be done with the map command using the functional 
operator x -> x - 1 as the first argument.  For example,

map(x -> x-1,[1,2,3,4,5,6]);
0, 1, 2, 3, 4, 5

The functional operator is applied to each element in the list given as the second argument. 

We need to apply the above to each composition in the set produced by the composition 
function.  To do this, we'll use a for loop over the result of composition and build a set of weak 
compositions.  Here is the complete procedure.

weakcomposition := proc(n::posint,r::posint)
  local minus1, strong, weak, C;
  minus1 := x -> x-1;
  strong := combinat[composition](n+r,r);
  weak := {};
  for C in strong do
    weak := weak union {map(minus1,C)};
  end do;
  return weak;
end proc:

The weak 2-compositions of 4 can now be produced by this procedure.
weakcomposition(4,2);

0, 4 , 1, 3 , 2, 2 , 3, 1 , 4, 0

Distinguishable Objects and Indistinguishable Boxes
As described in the text, the number of ways to place n distinguishable objects in k indistinguishable
boxes is given by the Stirling numbers of the second kind, S n, k .

The command stirling2 computes the Stirling number of the second kind.  This command 

requires two arguments, the number of objects and the number of boxes.  For example, the 
statement below computes the number of ways to put seven different employees in 4 different 
offices when each office must not be empty.

stirling2(7,4);
350

In order to compute the number of ways to assign the 7 employees to the 4 offices and allow empty 
offices, we must add the number of ways to assign all 7 employees to 1 office, to 2 offices, to 3 
offices, and to 4 offices.

add(stirling2(7,k),k=1..4);
715

Generating assignments of employees to offices
The stirling2 command tells us how many ways there are to place distinguishable objects in 
indistinguishable boxes, but Maple does not have a command to produce them. 

To make such a command, we rely on the following observations.  First, as indicated in the text, a 
choice of distinguishable objects to indistinguishable boxes can be modeled as a set of subsets.  For 
instance, A, C , D , B, E  represents the assignment of A and C to one box, D to a box of 
its own, and B and E to another box.  The set of subsets must not contain the empty set and must be 
such that the union of the subsets be the entire collection of objects.

We can produce such assignments recursively.  The basis step is that there is only one way to 
assign n objects to 1 box and there are no ways to assign n objects to k boxes for k O n (under the 
requirement that no box be empty.  To assign n objects to k boxes with k % n, proceed as follows.  

First, find all assignments of nK 1 objects to kK 1 boxes and update each assignment by placing 
object n in a box by itself.  In terms of the set representation, given B1, B2,…, Bk K 1 , we produce

B1, B2,…, Bk K 1, n .  

Second, find all assignments of nK 1 objects to k boxes.  For each such assignment 
B1, B2,…, Bk , produce the following k assignments of n objects to the k boxes:

B1W n , B2,…, Bk , B1, B2W n , B3,…, Bk ,…, B1, B2,…, BkW n .

The assignments produced by the two methods above produce all assignments.  The following 
procedure implements this algorithm.

makeStirling2 := proc(n::posint, k::posint)
  local A, k1boxes, kboxes, B, new, i;
  if k = 1 then
    return {{{$1..n}}};
  end if;
  if k > n then
    return {};
  end if;
  A := {};
  # n-1 objects in k-1 boxes
  k1boxes := makeStirling2(n-1,k-1);
  for B in k1boxes do
    new := B union {{n}};
    A := A union {new};
  end do;

  # n-1 objects in k boxes
  kboxes := makeStirling2(n-1,k);
  for B in kboxes do
    for i from 1 to k do
      new := subsop(i=B[i] union {n},B);
      A := A union {new};
    end do;
  end do;
  return A;
end proc:

Let us analyze the procedure.  It accepts n and k as parameters and returns the set consisting of all 
possible assignments of distinguishable objects to indistinguishable boxes.

In the case that k = 1, there is only one possible assignment, all objects are assigned to the single 
box.  This assignment is represented by 1, 2,…, n , since an assignment corresponds to a set of
subsets.  The procedure returns the set consisting of this single assignment when k = 1.  Recall the
$ operator with no left argument and a range as its right argument produces the sequence defined by
the range.

If k O n, then there are no valid assignments and the procedure returns the empty set.

Otherwise, we initialize A, which will store the set of assignments that is returned, to the empty set. 
Recall that there are two recursive steps.  First expanding on the assignments of nK 1 objects to 
kK 1 boxes.  And then expanding on the assignments of nK 1 objects to k boxes.

For the first part, we assign the name k1boxes to the set of assignments of nK 1 objects to kK 1 
boxes.  For each such assignment, e.g., 1, 3, 5 , 2 , 4, 6 , we add n in its own box:

1, 3, 5 , 2 , 4, 6 W 7 = 1, 3, 5 , 2 , 4, 6 , 7 .
This new assignment is then added to A.

In the second part, we assign kboxes to the set of assignments of nK 1 objects to k boxes.  For 
each such assignment, we consider each of the k boxes in turn and add n to that box.  For instance, 
the assignment 2, 3 , 1 , 5 , 4, 6  would generate the four assignments:

2, 3, 7 , 1 , 5 , 4, 6  
2, 3 , 1, 7 , 5 , 4, 6  
2, 3 , 1 , 5, 7 , 4, 6  
2, 3 , 1 , 5 , 4, 6, 7 .

Recall that the subsop command is used to replace (substitute) elements in sets and lists.  The 
syntax subsop(i=x,S) where i is an index of the set or list S, causes the expression x to 
replace whatever had been at index i in S.  

Compare the result of our procedure to the solution of Example 10 in Section 6.5 of the text.
makeStirling2(4,3);
1 , 2 , 3, 4 , 1 , 3 , 2, 4 , 1 , 4 , 2, 3 , 2 , 3 , 1, 4 ,

2 , 4 , 1, 3 , 3 , 4 , 1, 2
Except for using the integers 1 through 4 instead of the letters A through D, the output above is the 
same as the six ways listed in the text for placing the four employees in 3 offices.

To produce all 14 ways to assign the 4 employees to 3 offices with each office containing any 
number of employees, we need to loop over the different values of k.

offices := {}:
for k from 1 to 3 do
  offices := offices union makeStirling2(4,k):
end do:

We print them out one at a time.
for o in offices do
  print(o);
end do;

1, 2, 3, 4
1 , 2, 3, 4
2 , 1, 3, 4
3 , 1, 2, 4
4 , 1, 2, 3
1, 2 , 3, 4
1, 3 , 2, 4
1, 4 , 2, 3

1 , 2 , 3, 4
1 , 3 , 2, 4
1 , 4 , 2, 3
2 , 3 , 1, 4
2 , 4 , 1, 3
3 , 4 , 1, 2

Indistinguishable Objects and Indistinguishable Boxes
As described in the text, distributing n indistinguishable objects into k indistinguishable boxes is 
identical to forming a partition of n into k positive integers.  A partition of n into j positive integers 
is a sum n = a1 C a2 C/C aj with 0 ! a1 % a2 %/% aj.  (Note that this is the reverse order 
from the definition in the text.  Maple's commands for producing partitions list them in non-
decreasing order, rather than in non-increasing order as is done in the text.)

The Maple commands numbpart and partition are used to count and form partitions of 
integers.

With one argument, a nonnegative integer n, numbpart returns the total number of partitions of n 
into as many as n boxes.  Likewise, partition applied to one argument returns a list containing 
lists representing partitions of the argument.

For example, the statements below compute the number of partitions of 7 and lists all the partitions 
of 7.

numbpart(7);
15

partition(7);
1, 1, 1, 1, 1, 1, 1 , 1, 1, 1, 1, 1, 2 , 1, 1, 1, 2, 2 , 1, 2, 2, 2 , 1, 1, 1, 1, 3 , 1, 1, 2,

3 , 2, 2, 3 , 1, 3, 3 , 1, 1, 1, 4 , 1, 2, 4 , 3, 4 , 1, 1, 5 , 2, 5 , 1, 6 , 7

Both commands also accept a second argument, which specifies the maximum integer allowed to 
appear in a partition.  For example, to answer the question: how many ways are there to distribute 7 
indistinguishable balls in up to 7 identical boxes when each box can hold at most 4 objects, we 
would give 4 as the second argument to numbpart.

numbpart(7,4);
11

And to list them we give the same arguments with partition.
partition(7,4);

1, 1, 1, 1, 1, 1, 1 , 1, 1, 1, 1, 1, 2 , 1, 1, 1, 2, 2 , 1, 2, 2, 2 , 1, 1, 1, 1, 3 , 1, 1, 2,
3 , 2, 2, 3 , 1, 3, 3 , 1, 1, 1, 4 , 1, 2, 4 , 3, 4

We will now see how to determine the number of ways to partition n into at most k boxes.

Limiting the number of boxes
Example 11 asks how many ways there are to pack six copies of the same book into four boxes.  
The way that numbpart and partition are described, with the second argument controlling the 
maximum value in the partition rather than the number of entries in the list, suggests that they do not
answer this question.  But as it turns out, they do.

In fact, the partitions of n with at most k objects in a box are in one-to-one correspondence with the 
partitions of n into at most k boxes.  To understand why, consider the partition 1, 2, 2, 3 .  

Think about stacking the 8 objects in boxes according to the partition 1, 2, 2, 3 .  (The diagram 
shown below is a Ferrers diagram.)

X

X

X

 

 X

X

X

X

X

Instead of thinking about the columns as the boxes, we can instead consider the rows as boxes.

X

X X X

X X X X

Now the 8 objects are contained in three boxes.  One box (the top row) has 1 object, another (the 
middle row) has 3 objects, and the last box (the bottom row) has 4 objects.  In other words, we've 
partitioned 8 as 1, 3, 4 .  These two partitions are said to be conjugate.  (Note that you can also 
think about forming the conjugate by rotating the original diagram by 90 degrees or reflecting it 
across its diagonal.)

We began with a partition whose maximum entry was 3 and found that its conjugate was a partition 
into 3 boxes.  It is always the case that the conjugate of a partition with maximum n is a partition 
into n boxes.  Moreover, this correspondence is one-to-one.  We leave it to the reader to prove these
facts.

The consequence of this discussion is that the number of ways to pack 6 copies of the same book 
into at most 4 identical boxes that can each hold all the books is the same as the number of ways to 
pack 6 copies of the same book into boxes that can hold at most 4 books each.  That is, the solution 
to Example 11 is given by the following computation.

numbpart(6,4);
9

To produce these 9 partitions, we will use the conjpart command.  This command accepts a 
partition as its only argument.  It returns the conjugate of that partition. Using the example 
1, 2, 2, 3  from above,

conjpart([1,2,2,3]);
1, 3, 4

The statement partition(6,4) will produce the list of all partitions of 6 with maximum value 4.
Applying conjpart to each of those partitions produces the list of all partitions of 6 into at most 4
boxes.  We use map with conjpart as the first argument and partition(6,4) as the second 
argument to apply conjpart to each partition.

map(conjpart,partition(6,4));
6 , 1, 5 , 2, 4 , 3, 3 , 1, 1, 4 , 1, 2, 3 , 2, 2, 2 , 1, 1, 1, 3 , 1, 1, 2, 2

Generating partitions
Here we will describe how to generate partitions.  

The first observation to make is that the partitions of n are identical to the partitions of n when each 
box can hold at most n objects.  In terms of the partition command, the following are identical.

partition(7);
1, 1, 1, 1, 1, 1, 1 , 1, 1, 1, 1, 1, 2 , 1, 1, 1, 2, 2 , 1, 2, 2, 2 , 1, 1, 1, 1, 3 , 1, 1, 2,

3 , 2, 2, 3 , 1, 3, 3 , 1, 1, 1, 4 , 1, 2, 4 , 3, 4 , 1, 1, 5 , 2, 5 , 1, 6 , 7
partition(7,7);

1, 1, 1, 1, 1, 1, 1 , 1, 1, 1, 1, 1, 2 , 1, 1, 1, 2, 2 , 1, 2, 2, 2 , 1, 1, 1, 1, 3 , 1, 1, 2,
3 , 2, 2, 3 , 1, 3, 3 , 1, 1, 1, 4 , 1, 2, 4 , 3, 4 , 1, 1, 5 , 2, 5 , 1, 6 , 7

We will describe how to form the partitions of n objects when each box can hold at most k objects 
recursively.  The basis cases are: when k = 1, there is only one partition of n, the partition consisting
of n 1s; when n % 0, there are no partitions.  

To determine the partitions of n when each box can hold at most k objects, proceed as follows.  
First, determine all partitions of n when each box can hold at most kK 1 objects.  These partitions 
are also partitions of n satisfying the requirement that each box holds at most k.

Second, provided that nK k O 0, determine all partitions of nK k when each box can hold at most 
k objects, and append k to each partition.  For example, with n = 7 and k = 3, we have nK k = 4 and
the partitions of 4 with each box holding at most 3 are: 

partition(4,3);



(6.133)(6.133)

(6.101)(6.101)

O O 

O O 

(6.130)(6.130)

(6.135)(6.135)

O O 

(6.64)(6.64)

O O 

O O 

(6.91)(6.91)

(6.111)(6.111)

O O 

(6.132)(6.132)

O O 

(6.136)(6.136)

O O 

(6.131)(6.131)
O O 

O O 

(6.134)(6.134)

O O 

• • 

1, 1, 1, 1 , 1, 1, 2 , 2, 2 , 1, 3
Therefore, 1, 1, 1, 1, 3 , 1, 1, 2, 3 , 2, 2, 3 , 1, 3, 3  are partitions of 7 with each box having at
most 3 objects.

Combining the partitions of n with each box holding at most kK 1 objects with the partitions of n 
formed by appending k to the partitions of nK k with each box holding at most k objects produces 
all partitions of n with each box holding at most k objects.  Setting k = n produces all partitions of n.

It is an exercise to implement this algorithm.

Maple commands for generating partitions sequentially
We have seen that the partition command will produce all partitions of a given integer.  Maple 
also has commands for producing partitions sequentially.  Like subsets and cartprod for 
sequentially computing the subsets of a set and the Cartesian product of sets, producing partitions 
sequentially rather than all at once can save time and memory.

Maple's procedures are based on a canonical ordering of partitions.  This ordering begins with the 
partition of all 1s and ends with the partition n .  In this ordering, a partition xk, xk K 1,…, x2, x1  
precedes the partition yj, yj K 1,…, y2, y1  if, for an index i, 

x1 = y1, x2 = y2, ..., xi K 1 = yi K 1, and xi ! yi.  
In other words, partition x precedes partition y when the first (from the right) position at which they 
differ contains a smaller value in x than in y.

For example, the partition 1, 1, 1, 3, 5  precedes the partition 1, 2, 3, 5  because they agree in the 
two right-most positions, but in the third position from the right, 1, 1, 1, 3, 5 's entry is smaller 
than the third value from the right in 1, 2, 3, 5 .  This is called reverse lexicographic order.

The commands firstpart and lastpart accept a positive integer n as their only argument.  
They return the partition consisting of n 1s and the partition n , respectively.

firstpart(11);
1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1

lastpart(11);
11

The command nextpart accepts as its sole argument a partition, that is, a non-decreasing list of 
positive integers.  It returns the next partition in the ordering.

nextpart([1,1,1,2,2,4]);
1, 2, 2, 2, 4

Likewise, prevpart accepts a partition and returns the previous partition in the ordering.
prevpart(%);

1, 1, 1, 2, 2, 4

Using these commands, we can easily generate all partitions of a given number one at a time.  We 
initialize a name with the firstpart command.  Inside a while loop controlled by testing the 
current partition against the last partition, we update the partition using the nextpart command.

p := firstpart(5);
p := 1, 1, 1, 1, 1

while p <> lastpart(5) do

  p := nextpart(p);
end do;

p := 1, 1, 1, 2
p := 1, 2, 2
p := 1, 1, 3
p := 2, 3
p := 1, 4
p := 5

Maple also includes the randpart command for generating a random partition.  This command 
accepts only one argument, the integer to be partitioned.

randpart(5);
1, 4

6.6 Generating Permutations and Combinations
In this section, we will implement Algorithm 1 from Section 6.6 of the text for generating the next 
permutation in lexicographic order.  Implementing Algorithms 2 and 3 will be left as exercises for 
the reader.  

Interchange
Before implementing Algorithm 1, we will first write a procedure to interchange two elements in a 
list.  This will be called by the procedure for generating permutations.

The Interchange procedure will require three parameters: the list and two integers representing 
the indices to be swapped.

The procedure has five statements.  A copy of the list is made, since parameters cannot be assigned 
to.  The element in the list at the first position to be swapped is assigned to a local variable.  Then 
the first position is assigned to the value in the second position.  Finally, the second position is 
assigned to the value stored in the temporary name and the list is returned.

Here is the implementation and an example of applying it.
Interchange := proc(L::list, i::integer, j::integer)
  local l, temp;
  l := L;
  temp := l[i];
  l[i] := l[j];
  l[j] := temp;
  return l;
end proc:
Interchange(["a","b","c","d","e","d"],2,5);

"a", "e", "c", "d", "b", "d"

nextperm
The input to the procedure will be a permutation a1, a2,…, an  of the set 1, 2,…, n .  Algorithm 
1 consists of three steps: finding the largest j such that aj ! aj C 1; finding the smallest ak to the 
right of aj and interchanging ak and aj; and putting the elements in positions jC 1 and beyond in 
increasing order.

The first step comprises the first four lines of the body of Algorithm 1, through the comment.  The 
index j is initialized to the next to last index in the permutation.  A while loop is used to conduct the 
search.  The body of the while loop decreases the value of j by one, and it is controlled by the 
condition aj O aj C 1.  When the while loop terminates, it will be the case that aj ! aj C 1 and j is 
the largest index for which that is true.  Consequently, aj C 1 O aj C 2 O/O an.

The second step is to find the smallest ak to the right of and larger than aj and interchange the two.  
Since we are guaranteed that the elements to the right of aj are in increasing order, an is the smallest 
element to the right of aj, an K 1 is the next smallest, and so on.  We are again searching from the 
right.  Initialize k to n.  A while loop is used to decrease k by one so long as aj O ak.  When the 
while loop terminates, k will be such that aj ! ak.  Note that the loop is guaranteed to stop with 
k O j since aj ! aj C 1.  Once j has been identified, we interchange aj and ak using the
Interchange procedure.

The third step is to put the elements of the permutation to the right of position j in increasing order.  
Note that before the interchange, aj C 1 through an were in decreasing order.  After the interchange 
of aj with ak, the tail end of the permutation remains in decreasing order.  This is because aj was 
smaller than ak, but ak was the smallest of the entries bigger than aj.  Thus all of ak C 1,…, an are 
smaller than aj, and all of aj C 1,…, ak K 1 are larger than ak which is larger than aj.  Therefore,

aj C 1,…, ak K 1, aj, ak C 1,…, an
is in decreasing order.

To put the tail in increasing order, we follow the instructions in the pseudocode that follow the 
interchange of aj and ak.  Variables r and s are initialized to n and jC 1, respectively.  Provided that
r remains larger than s, we interchange ar and as and then decrease r by 1 and increase s by 1.  This
has the effect of swapping aj C 1 with an, then aj C 2 with an K 1, then aj C 3 with an K 2, etc.  

Once the tail is in increasing order, the result is the new permutation and it is returned.

We need to add to the procedure two tests to ensure that the input is valid.  We will declare the input
to be a list of positive integers.  Within the body of the procedure, we will check first to ensure that 
all of the elements in the list are no greater than n, the length of the list.  Otherwise, the procedure 
will generate an error.  We will also compare the input to the final permutation n, nK 1,…, 1  and 
return FAIL if they are equal.

Here is the implementation.
nextperm := proc(A::list(posint))
  local a, n, i, j, k, r, s;
  a := A;
  n := nops(a);
  for i from 1 to n do
    if a[i] > n then 
      error "Input must be a permutation of {1,2,...,n}.";
    end if;
  end do;
  if a = [seq(n-i,i=0..(n-1))] then
    return FAIL;

  end if;
  # step 1: find j
  j := n - 1;
  while a[j] > a[j+1] do
    j := j - 1;
  end do;
  # step 2: find k and interchange aj and ak
  k := n;
  while a[j] > a[k] do
    k := k - 1;
  end do;
  a := Interchange(a,j,k);
  # step 3: sort the tail
  r := n;
  s := j + 1;
  while r > s do
    a := Interchange(a,r,s);
    r := r - 1;
    s := s + 1;
  end do;
  return a;
end proc:

Example 2 of Section 6.6 finds that the permutation after 362541 is 364125.  Let's use that example 
to confirm that our procedure is working.

nextperm([3,6,2,5,4,1]);
3, 6, 4, 1, 2, 5

To generate all permutations of a set 1, 2,…, n , we use a while loop.
aperm := [$1..4]:
while aperm <> FAIL do
  print(aperm);
  aperm := nextperm(aperm):
end do:

1, 2, 3, 4
1, 2, 4, 3
1, 3, 2, 4
1, 3, 4, 2
1, 4, 2, 3
1, 4, 3, 2
2, 1, 3, 4
2, 1, 4, 3
2, 3, 1, 4
2, 3, 4, 1
2, 4, 1, 3
2, 4, 3, 1
3, 1, 2, 4
3, 1, 4, 2

3, 2, 1, 4
3, 2, 4, 1
3, 4, 1, 2
3, 4, 2, 1
4, 1, 2, 3
4, 1, 3, 2
4, 2, 1, 3
4, 2, 3, 1
4, 3, 1, 2
4, 3, 2, 1

Finding Permutations of Other Sets
As mentioned in the text, any set with n elements can be put in one-to-one correspondence with 

1, 2,…, n .  Consequently, any permutation of a set with n elements can be obtained from a 
permutation of 1, 2,…, n  and the correspondence.

In Maple, we can use the subs command to transform a permutation of 1, 2,…, n  into a 
permutation of another set with n elements.

The subs command is used to perform substitutions in an expression.  It has several forms, but we
will use it with two arguments.  The second argument will be the permutation of 1, 2,…, n .  The 
first argument will be a list of equations of the form i = x where i is an integer in 1, 2,…, n  and x 
is the corresponding element of the set being permuted.

For example, consider the set a, b, c  and the permutation 3, 1, 2  of 1, 2, 3 .  We establish the 
correspondence that 1, 2, and 3 are identified with a, b, and c, respectively.  To produce the 
corresponding partition, we use the subs command with first argument 1 = a, 2 = b, 3 = c  to 
specify the correspondence.  The second argument is the permutation 3, 1, 2 .

subs([1="a",2="b",3="c"],[3,1,2]);
"c", "a", "b"

As another example, consider the set 2, 10, 13, 19  and the permutation 4, 2, 3, 1 .  Identify 1 
with 2, 2 with 10, 3 with 13, and 4 with 19.  Then the following command produces the desired 
permutation.

subs([1=2,2=10,3=13,4=19],[4,2,3,1]);
19, 10, 13, 2

Note that subs will allow you to list the equations without enclosing them in a list.  However, 
putting them in a list ensures that the substitutions are done simultaneously rather than sequentially. 
In this example, sequential assignment would not have produced the correct result.

A procedure to apply permutations
We conclude by writing a procedure that, given a set and a permutation of 1, 2,…, n  with n equal 
to the size of the set, will return the corresponding permutation of the set.

The key is automating the creation of the equations defining the correspondence.  This can be done 
with the zip command.  Recall that zip requires three arguments.  The second and third arguments
are lists while the first argument is a procedure on two parameters.  The result is the list formed by 



(6.101)(6.101)

O O 

O O 

O O 

(6.130)(6.130)

O O 

(6.144)(6.144)

O O 

(6.64)(6.64)

O O 

O O 

(6.148)(6.148)

O O 

(6.91)(6.91)

O O 
(6.147)(6.147)

(6.111)(6.111)

O O 

(6.146)(6.146)

(6.145)(6.145)

O O 

O O 

O O 

(6.136)(6.136)

(6.140)(6.140)

O O 

• • 

(6.143)(6.143)

applying the procedure to corresponding pairs of elements from the two lists.   

We create a functional operator that takes two expressions and forms an equation from them.
makeEqn := (a,b) -> a=b;

makeEqn := a, b /a = b
makeEqn(5,a);

5 = a

Using the list 1, 2, 3  and the list formed from the elements of the set a, b, c  as the second and 
third arguments, zip will produce the list of equations used in (6.141).

zip(makeEqn,[1,2,3],["a","b","c"]);
1 = "a", 2 = "b", 3 = "c"

This is all we need to write a procedure that accepts a set and a permutation of 1, 2,…, n  and 
returns the corresponding permutation of the elements of the set.

permuteSet := proc(S::set,P::list(posint))
  local L, M, makeeqn, a, b, eqns;
  L := [op(S)];
  M := [$1..nops(S)];
  makeeqn := (a,b) -> a=b;
  eqns := zip(makeeqn,M,L);
  return subs(eqns,P);
end proc:

Using the procedure, we can compute the permutation of a, b, c, d, e  corresponding to 
3, 5, 2, 1, 4 .

permuteSet({"a","b","c","d","e"},[3,5,2,1,4]);
"c", "e", "b", "a", "d"

Solutions to Computer Projects and Computations and Explorations
Computer Projects 10

Given positive integers n and r, list all the r-combinations, with repetition allowed, of the set 
1, 2, 3,…, n .

Solution: In Section 6.5 of this manual, we showed that the choose command could be used to 
generate combinations with repetition by repeating the elements in the list given as the first 
argument to choose.  

To generate the 2-combinations of 1, 2, 3 , for example, we apply the choose command to 
the list consisting of 1, 2, and 3, each repeated twice.  Note that the number of repetitions must 
be the same as r in order to choose r all of the same object.

choose([1$2, 2$2, 3$2],2);
1, 1 , 1, 2 , 1, 3 , 2, 2 , 2, 3 , 3, 3

Recall that x$n generates the sequence of n x's.

The first argument to choose can be generated using seq as follows.
[seq(i$2,i=1..3)];

1, 1, 2, 2, 3, 3

We now write a procedure that accepts n and r as input and produces all r-combinations with 
repetition allowed.

chooseRepetition := proc(n::posint, r::posint)
  local L, i;
  L := [seq(i$r,i=1..n)];
  return combinat[choose](L,r);
end proc:

We can obtain all of the 3-combinations of 1, 2, 3, 4, 5  by
chooseRepetition(5,3);

1, 1, 1 , 1, 1, 2 , 1, 1, 3 , 1, 1, 4 , 1, 1, 5 , 1, 2, 2 , 1, 2, 3 , 1, 2, 4 , 1, 2,
5 , 1, 3, 3 , 1, 3, 4 , 1, 3, 5 , 1, 4, 4 , 1, 4, 5 , 1, 5, 5 , 2, 2, 2 , 2, 2, 3 ,
2, 2, 4 , 2, 2, 5 , 2, 3, 3 , 2, 3, 4 , 2, 3, 5 , 2, 4, 4 , 2, 4, 5 , 2, 5, 5 ,
3, 3, 3 , 3, 3, 4 , 3, 3, 5 , 3, 4, 4 , 3, 4, 5 , 3, 5, 5 , 4, 4, 4 , 4, 4, 5 ,
4, 5, 5 , 5, 5, 5

Computations and Explorations 1

Find the number of possible outcomes in a two-team playoff when the winner is the first team 
to win 5 out of 9, 6 out of 11, 7 out of 13, and 8 out of 15.

Solution: We will designate the two teams as 1 and 2 and model a playoff as a list of 1s and 2s.  
For example, 1, 2, 2, 1, 1, 1, 2, 1  is a playoff in which team 1 wins the first game, team 2 wins
games 2 and 3, team 1 wins games 4, 5, and 6, team 2 wins game 7, and them team 1 wins game
8.  If the winner is the first team to win 5 out of 9 games, then team 1 has won the tournament 
after 8 games.

We will write a procedure to produce all of the possible outcomes in a playoff where the winner 
is the first team to win n out of 2 nK 1 games.

First we will create a small procedure that will determine, given a list of the outcomes of 
individual games and the number of games needed to win, whether a team has won the playoff 
or not.  The procedure will return true if one of the teams has won or false if neither team has 
reached the threshold for winning.

The procedure counts the 1s and 2s in the list.  If either number is equal to n, it returns true. 
playoffWon := proc(L::list({1,2}),n::posint)
  local n1, n2, i;
  n1 := 0;
  n2 := 0;
  for i from 1 to nops(L) do
    if L[i] = 1 then
      n1 := n1 + 1;
    else
      n2 := n2 + 1;
    end if;
  end do;
  if n1 = n or n2 = n then
    return true;
  else
    return false;
  end if;

end proc:
For instance, in our example 1, 2, 2, 1, 1, 1, 2, 1 , the procedure recognizes that the playoff has
been won.

playoffWon([1,2,2,1,1,1,2,1],5);
true

We will construct the possible outcomes as follows.  Begin with a set outcomes and a list S.  
Initialize outcomes to the empty set and S to the list 1 , 2 .  

Consider the first element of S, say p.  Remove p from the list.  Then construct the two lists 
formed by adding 1 and 2 respectively to p.  For each of these, use playoffWon to determine 
whether or not they are outcomes.  If so, they are added to the outcomes set, and if not, they 
are added to the end of S.  When S is empty, then outcomes consists of all possible outcomes 
of the playoff.

Here is the procedure.
allPlayoffs := proc(n::posint)
  local outcomes, S, p, p1, p2;
  outcomes := {};
  S := [[1],[2]];
  while S <> [] do
    p := S[1];
    S := S[2..-1];
    p1 := [op(p),1];
    p2 := [op(p),2];
    if playoffWon(p1,n) then
      outcomes := outcomes union {p1};
    else
      S := [op(S),p1];
    end if;
    if playoffWon(p2,n) then
      outcomes := outcomes union {p2};
    else
      S := [op(S),p2];
    end if;
  end do;
  return outcomes;
end proc:

We now apply this procedure to playoffs that are best 3 out of 5.  
best3of5 := allPlayoffs(3);

best3of5 := 1, 1, 1 , 2, 2, 2 , 1, 1, 2, 1 , 1, 2, 1, 1 , 1, 2, 2, 2 , 2, 1, 1, 1 , 2,
1, 2, 2 , 2, 2, 1, 2 , 1, 1, 2, 2, 1 , 1, 1, 2, 2, 2 , 1, 2, 1, 2, 1 , 1, 2, 1, 2, 2 ,
1, 2, 2, 1, 1 , 1, 2, 2, 1, 2 , 2, 1, 1, 2, 1 , 2, 1, 1, 2, 2 , 2, 1, 2, 1, 1 , 2, 1,

2, 1, 2 , 2, 2, 1, 1, 1 , 2, 2, 1, 1, 2
nops(best3of5);

20
The reader is left to apply the procedure to the cases called for in the problem and to conjecture a 
general formula.

Computations and Explorations 3

Verify that C 2 n, n  is divisible by the square of a prime, when n s 1, 2, or 4, for as many 
positive integers n as you can.  [The theorem that tells that C 2 n, n  is divisible by the square
of a prime for n s 1, 2, or 4 was proved in 1996 by Andrew Granville and Olivier Ramaré.  
Their proof settled a conjecture made in 1980 by Paul Erdős and Ron Graham.]

Solution:  We will first consider one example to see exactly what we need to do.  Then we will 
write a general procedure.  Consider n = 3, the smallest n for which the theorem is true.

First, compute C 2 n, n  for n = 3.
c := binomial(6,3);

c := 20

To determine whether or not C 2 n, n  is divisible by the square of a prime, we'll look at its 
prime factorization.  If any of the exponents in the prime factorization are 2 or greater, then we 
know the number is divisible by the square of the corresponding prime.

We use the command ifactors (first discussed in Section 4.3 of this manual).  The
ifactors command requires one argument, an integer.  It's output is a list of the form 
sign, p1, e1 , p2, e2 ,…, pm, em  where sign is positive or negative 1; p1, p2,…, pm are 

the primes in the prime factorization; and e1, e2,…, em are the corresponding exponents.

Apply ifactors to C 6, 3 .
ifactors(c);

1, 2, 2 , 5, 1

The result tells us that C 6, 3 = 1$22$51.

We are interested in the exponents of the primes.  We take the second element of the list to obtain
the list of pairs of primes and their exponents without the sign component.

cFacts := ifactors(c)[2];
cFacts := 2, 2 , 5, 1

This gives us a list of lists, where the first element in each internal list is the prime factor and the 
second element in each pair is the exponent associated to that prime.  Since we are only 
interested in the exponents, we'll separate them out.

cPowers := seq(x[2],x=cFacts);
cPowers := 2, 1

Provided that there is an exponent of 2 or larger, we know that the square of a prime divides the 
number.  In this case, the first exponent we obtained is 2, so in fact C 6, 3  is divisible by the 
square of a prime.

Combining these steps, we write a procedure, HasPrimeSquare, that, given n, returns true if 
C 2 n, n  is divisible by the square of a prime and false if not.

HasPrimeSquare := proc(n::posint)
  local c, facts, powers, x, e;
  c := binomial(2*n,n);
  facts := ifactors(c)[2];
  powers := seq(x[2],x=facts);
  for e in powers do
    if e >= 2 then

      return true;
    end if;
  end do;
  return false;
end proc:

Now we write a second routine that will test the values of n from 1 to a maximum.  The 
procedure prints a message whenever it finds an n for which HasPrimeSquare returns false.  
It also prints a message when it concludes.

CheckPrimeSquare := proc(m::posint)
  local n;
  for n from 1 to m do
    if not HasPrimeSquare(n) then
      print("Found counterexample: ",n);
    end if;
  end do;
  print("finished");
end proc:

Applying this procedure to 100 verifies that the only values of n for which C 2 n, n  is not 
divisible by the square of a prime are 1, 2, and 4, as expected.

CheckPrimeSquare(100);
"Found counterexample: ", 1
"Found counterexample: ", 2
"Found counterexample: ", 4

"finished"

Exercises
Exercise 1.  Build a recursive version of SubsetSumCount, using the ideas of
FindBitStrings.  Your procedure should determine all subsets of a given set whose sum is 
less than a target value.  Rather than considering all sets, it should build potential sets recursively 
using the fact that once a set has sum larger than the target no larger set of positive integers can have
smaller sum.  Compare the performance of your procedure with SubsetSumCount. 

Exercise 2.  Create a procedure FindDecreasing by modifying FindIncreasing in order to 
determine a strictly decreasing subsequence of maximal length.

Exercise 3.  Modify the Patience algorithm to find all of the strictly increasing subsequences of 
maximal length.

Exercise 4.  Use Maple to show that, in Example 11, n positive integers not exceeding 2 n are not 
sufficient to guarantee that one integer divides one of the others.

Exercise 5.  Use Maple to determine how many different strings can be made from the word 
"PAPARAZZI" when all the letters are used, when any number of letters are used, when all the 
letters are used and the string begins and ends with the letter "Z", and when all the letters are used 
and the three "A"s are consecutive.

Exercise 6.  Suppose that a certain Mathematics Department has m male faculty and f female 
faculty. Write a Maple procedure to find all committees with 2 k members in which both sexes are 
represented equally.

Exercise 7.  Use Maple to prove the identity
nC 1

k =
nC 1

k
n

kK 1
for positive integers n and k with k % n.

Exercise 8.  Use Maple to prove Pascal's identity:
nC 1

k =
n

kK 1 C
n
k

for all positive integers n and k with k % n.

Exercise 9.  Use Maple to generate many rows of Pascal's triangle.  See if you can formulate any 
conjectures involving identities satisfied by the binomial coefficients.  Use Maple to help you verify 
that your conjecture is true by using the techniques at the end of Section 6.4 of this manual.

Exercise 10.  Write a procedure that mixes the techniques used in BinomialF and BinomialR to
generate the rows of Pascal's triangle from row a to row b for b O a O 0.  

Exercise 11.  Use Maple to count and list all solutions to the equation
x1 C x2 C x3 C x4 = 25,

where x1, x2, x3, and x4 are non-negative integers.  Also count and list all solutions such that 
x1 R 1, x2 R 2, x3 R 3, and x4 R 4.

Exercise 12.  Generate a large triangle of Stirling numbers of the second kind and look for patterns 
that suggest identities among the Stirling numbers.  Also see if you can make any conjectures about 
the relationship between Stirling numbers and the binomial coefficients.

Exercise 13.  Implement the algorithm described in the Generating partitions subsection of Section 
6.5 of this manual.

Exercise 14.  Write a Maple procedure that takes as input three positive integers n, k, and i, and 
returns the ith multinomial, in lexicographic order, of the polynomial x1 C x2 C/C xk

n.  Write 
its inverse; that is, given a multinomial, the inverse should return its index (position) in the sorted 
polynomial.

Exercise 15.  Implement Algorithm 2 of Section 6.6 for generating the next largest bit string.

Exercise 16.  Implement Algorithm 3 of Section 6.6 for generating the next r-combination.

Exercise 17.  Write a Maple procedure to compute the Cantor expansion of an integer.  (See the 
prelude to Exercise 14 of Section 6.6 of the text.)

Exercise 18.  Implement the algorithm for generating the set of all permutations of the first n 
integers using the bijection from the collection of all permutations of the set 1, 2,…, n  to the set 

1, 2,…, n!  described prior to Exercise 14 of Section 6.6 of the textbook.


